PRÁCTICAS DE ÁLGEBRA Y MATEMÁTICA DISCRETA.
BASES Y COORDENADAS

```
latex.matrix_delimiters("[", "]")
```

Ejercicio 1. *Sea V el espacio vectorial cuya base son las columnas de la matriz*

$$B = \begin{bmatrix} 0 & 2 & 2 & 2 & 3 & 1 \\ 4 & 0 & 4 & 6 & 6 & 6 \\ 6 & 1 & 5 & 1 & 2 & 3 \\ 4 & 3 & 0 & 2 & 6 & 1 \\ 2 & 6 & 6 & 4 & 4 & 0 \\ 4 & 3 & 5 & 3 & 5 & 5 \end{bmatrix} \in \mathbf{M}_{6\times 6}(\mathbb{Z}_7)$$

Determina si el vector $y = \begin{bmatrix} 2 \\ 2 \\ 1 \\ 5 \\ 0 \\ 4 \end{bmatrix}$ *está en V y en caso de estarlo, calcula sus coordenadas en base B.*

Solución:

```
B=matrix(Zmod(7),[[0,2,2,2,3,1],[4,0,4,6,6,6],
[6,1,5,1,2,3],[4,3,0,2,6,1],
[2,6,6,4,4,0],[4,3,5,3,5,5]])
y=column_matrix(Zmod(7),[2,2,1,5,0,4])
By = block_matrix(1,2,[B,y])
R=By.echelon_form()
```

El vector y está en $V = C(B)$ si es combinación lineal de las columnas de la matriz B. Ello equivale a que el sistema de ecuaciones $B \cdot x = y$ es compatible.

Determinamos la matriz ampliada

$$\left[\begin{array}{cccccc|c} 0 & 2 & 2 & 2 & 3 & 1 & 2 \\ 4 & 0 & 4 & 6 & 6 & 6 & 2 \\ 6 & 1 & 5 & 1 & 2 & 3 & 1 \\ 4 & 3 & 0 & 2 & 6 & 1 & 5 \\ 2 & 6 & 6 & 4 & 4 & 0 & 0 \\ 4 & 3 & 5 & 3 & 5 & 5 & 4 \end{array}\right]$$

Reducimos por filas

$$\left[\begin{array}{cccccc|c} 1 & 0 & 0 & 0 & 0 & 0 & 4 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 5 \\ 0 & 0 & 0 & 1 & 0 & 0 & 2 \\ 0 & 0 & 0 & 0 & 1 & 0 & 6 \\ 0 & 0 & 0 & 0 & 0 & 1 & 5 \end{array}\right]$$

Como la columna de términos independientes no es pivote, el sistema es compatible y el vector y está en V. Además se tiene que

$$y = \begin{bmatrix} 2 \\ 2 \\ 1 \\ 5 \\ 0 \\ 4 \end{bmatrix} = \begin{bmatrix} 4 \\ 0 \\ 5 \\ 2 \\ 6 \\ 5 \end{bmatrix}_B$$

Ejercicio 2. *Sea V el espacio vectorial cuya base son las columnas de la matriz*

$$B = \begin{bmatrix} 1 & -2 & -5 \\ 0 & 1 & 3 \\ 3 & -4 & -8 \\ -1 & 0 & 4 \\ 1 & -4 & -6 \end{bmatrix} \in \mathbf{M}_{5\times 3}(\mathbb{R})$$

Determina si el vector $y = \begin{bmatrix} -5 \\ 3 \\ -8 \\ 4 \\ -5 \end{bmatrix}$ *está en V y en caso de estarlo, calcula sus coordenadas en base B.*

Solución:

Ejercicio 3. *Sea V un espacio vectorial del cual conocemos dos bases, B y B' que corresponden a las columnas de las siguientes matrices*

$$B' = \begin{bmatrix} 15 & -8 & -38 \\ 2 & -1 & -6 \\ -10 & 5 & 31 \\ 4 & -2 & -14 \\ 8 & -4 & -23 \\ 0 & 0 & -3 \end{bmatrix}, B = \begin{bmatrix} 1 & 7 & -1 \\ 0 & 1 & -1 \\ 0 & -5 & 6 \\ 0 & 2 & -4 \\ 0 & 4 & -3 \\ 0 & 0 & -3 \end{bmatrix} \in \mathbf{M}_{6\times 3}(\mathbb{R})$$

Calcula la matriz de cambio de base $P_{B'B}$.

Solución:

Dadas las bases B y B' sabemos que la matriz del cambio $P_{B'B}$ es la matriz $A \cdot B'$ donde A es una inversa lateral por la izquierda de B.

```
B=matrix(QQ,[[1,7,-1],[0,1,-1],
[0,-5,6],[0,2,-4],
[0,4,-3],[0,0,-3]])
B1=matrix(QQ,[[15,-8,-38],[2,-1,-6],
[-10,5,31],[4,-2,-14],
[8,-4,-23],[0,0,-3]])
BI = block_matrix(1,2,[B,1])
R=BI.echelon_form()
R=copy(R)
R.subdivide(3,3)
A=R.subdivision(0,1)
```

Pasamos a determinar una inversa lateral por la izquierda de la matriz B. Para ello ampliamos a la derecha de la matriz B con la identidad

$$\left[\begin{array}{rrr|rrrrrr} 1 & 7 & -1 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & -1 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & -5 & 6 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 2 & -4 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 4 & -3 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & -3 & 0 & 0 & 0 & 0 & 0 & 1 \end{array}\right]$$

Reducimos por filas dicha matriz

$$\left[\begin{array}{rrr|rrrrrr} 1 & 0 & 0 & 1 & 0 & 0 & 0 & -\frac{7}{4} & \frac{17}{12} \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & \frac{1}{4} & -\frac{1}{4} \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & -\frac{1}{3} \\ \hline 0 & 0 & 0 & 0 & 1 & 0 & 0 & -\frac{1}{4} & -\frac{1}{12} \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & \frac{5}{4} & \frac{3}{4} \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & -\frac{1}{2} & -\frac{5}{6} \end{array}\right]$$

Una inversa lateral por la izquierda de B es la matriz A, donde

$$A = \begin{bmatrix} 1 & 0 & 0 & 0 & -\frac{7}{4} & \frac{17}{12} \\ 0 & 0 & 0 & 0 & \frac{1}{4} & -\frac{1}{4} \\ 0 & 0 & 0 & 0 & 0 & -\frac{1}{3} \end{bmatrix}$$

Por tanto

$$P_{B'B} = \begin{bmatrix} 1 & 0 & 0 & 0 & -\frac{7}{4} & \frac{17}{12} \\ 0 & 0 & 0 & 0 & \frac{1}{4} & -\frac{1}{4} \\ 0 & 0 & 0 & 0 & 0 & -\frac{1}{3} \end{bmatrix} \cdot \begin{bmatrix} 15 & -8 & -38 \\ 2 & -1 & -6 \\ -10 & 5 & 31 \\ 4 & -2 & -14 \\ 8 & -4 & -23 \\ 0 & 0 & -3 \end{bmatrix} = \begin{bmatrix} 1 & -1 & -2 \\ 2 & -1 & -5 \\ 0 & 0 & 1 \end{bmatrix}$$

Este problema es equivalente a calcular la matriz de la aplicación lineal

$$V_{B'} \xrightarrow{id} V_B$$

y podemos aplicar la fórmula

$$P_{B'B} = M_{B'B}(id) = A\ id(B') = AB' = \begin{bmatrix} 1 & -1 & -2 \\ 2 & -1 & -5 \\ 0 & 0 & 1 \end{bmatrix}.$$

Ejercicio 4. *Sea V un espacio vectorial del cual conocemos la base B que corresponden a las columnas de la matriz*

$$B = \begin{bmatrix} 1 & 7 & -1 & 2 \\ 0 & 1 & -1 & -1 \\ 0 & -5 & 6 & -4 \\ 0 & 2 & -4 & -1 \end{bmatrix} \in \mathbf{M}_{4\times 4}(\mathbb{Z}_7)$$

Calcula la matriz de cambio de base $P_{C_4 B}$.

Solución:

```
B=matrix(Zmod(7),[[1,7,-1,2],[0,1,-1,-1],
[0,-5,6,-4],[0,2,-4,-1]])
```

Tenemos el siguiente diagrama:

$$V_{C_4} \xrightarrow{id} V_B$$

En general, la matriz del cambio de base es la inversa por la izquierda de B por la matriz de C_4. Como B es cuadrada, su inversa por la izquierda es su inversa normal y la matriz asociada a la base canónica es la identidad, por tanto

$$P_{C_4 B} = B^{-1} I = \begin{bmatrix} 1 & 5 & 1 & 0 \\ 0 & 5 & 6 & 6 \\ 0 & 2 & 2 & 4 \\ 0 & 2 & 4 & 2 \end{bmatrix}.$$

Ejercicio 5. *Sea la aplicación* $f : \mathbb{R}^3 \to \mathbb{R}^3$ *tal que*

$$M(f) = \begin{bmatrix} 1 & 2 & 3 \\ -3 & -5 & -8 \\ -1 & -5 & -6 \end{bmatrix}.$$

Si B' *y* B *son bases dadas por las matrices*

$$B' = \begin{bmatrix} 1 & 1 & 6 \\ -1 & 0 & -2 \\ 0 & 0 & 1 \end{bmatrix} \text{ y } B = \begin{bmatrix} -2 & 7 & 3 \\ 1 & -4 & -1 \\ 0 & 0 & 1 \end{bmatrix},$$

Calcula $M_{B'B}(f)$.

Solución:

```
Mf = matrix(QQ,[[1,2,3],[-3,-5,-8],[-1,-5,-6]])
B1=matrix(QQ,[[1,1,6],[-1,0,-2],[0,0,1]])
B=matrix(QQ,[[-2,7,3],[1,-4,-1],[0,0,1]])
```

Consideremos el siguiente diagrama:

$$\begin{tikzcd} \mathbb{R}^3_{B'} \arrow[r, "id", "M_{B'C}(id)"'] \arrow[rrr, bend right, "f", "M_{B'B}(f)"'] & \mathbb{R}^3_C \arrow[r, "f", "M_{CC}(f)"'] & \mathbb{R}^3_C \arrow[r, "id", "M_{CB}(id)"'] & \mathbb{R}^3_B \end{tikzcd}$$

Si calculamos cada una de las matrices de esas aplicaciones tenemos que

- $M_{CB}(id) = P_{CB} = B^{-1} I = B^{-1}$ porque B es cuadrada y su inversa por la izquierda es su inversa.
- $M_{CC}(f) = M(f)$ nos la dan.
- $M_{B'C}(id) = P_{B'C} = I^{-1} B' = B'$ porque la matriz asociada a la base canónica es la matriz identidad, que es cuadrada y su inversa es ella misma.

Por lo tanto

$$M_{B'B}(f) = B^{-1} M(f) B' =$$

$$\begin{bmatrix} -4 & -7 & 5 \\ -1 & -2 & 1 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 2 & 3 \\ -3 & -5 & -8 \\ -1 & -5 & -6 \end{bmatrix} \begin{bmatrix} 1 & 1 & 6 \\ -1 & 0 & -2 \\ 0 & 0 & 1 \end{bmatrix} = \begin{bmatrix} 10 & 12 & 82 \\ 1 & 4 & 25 \\ 4 & -1 & -2 \end{bmatrix}.$$

Ejercicio 6. *Sea la aplicación* $f : \mathbb{R}^4 \to \mathbb{R}^3$ *tal que*

$$M_{B'B}(f) = \begin{bmatrix} -2 & -6 & 2 & 0 \\ -3 & -9 & 3 & 0 \end{bmatrix}$$

siendo B' *y* B *las bases dadas por las matrices*

$$B' = \begin{bmatrix} 6 & -4 & 0 & 3 \\ -1 & 2 & -4 & -7 \\ 0 & 2 & -5 & -7 \\ 5 & -3 & -1 & 1 \end{bmatrix} \text{ y } B = \begin{bmatrix} -1 & 5 \\ -2 & 9 \end{bmatrix},$$

Calcula $M(f)$.

Solución:

```
MB1Bf = matrix(QQ,[[-2,-6,2,0],[-3,-9,3,0]])
B1=matrix(QQ,[[6,-4,0,3],[-1,2,-4,-7],[0,2,-5,-7],[5,-3,-1,1]])
B=matrix(QQ,[[-1,5],[-2,9]])
```

Consideremos el siguiente diagrama:

$$\begin{tikzcd}
\mathbb{R}^4_C \arrow[r, "id", "M_{CB'}(id)"'] \arrow[rrr, bend right, "f", "M_{CC}(f)"'] & \mathbb{R}^4_{B'} \arrow[r, "f", "M_{B'B}(f)"'] & \mathbb{R}^2_B \arrow[r, "id", "M_{BC}(id)"'] & \mathbb{R}^2_C
\end{tikzcd}$$

Si calculamos cada una de las matrices de esas aplicaciones tenemos que

- $M_{CB'}(id) = P_{CB'} = (B')^{-1}I = (B')^{-1}$.
- $M_{B'B}(f)$ nos la dan.
- $M_{BC}(id) = P_{BC} = I^{-1}B = B$.

De donde

$$M(f) = M_{CC}(f) = M_{BC}(id)\ M_{B'B}(f)\ M_{CB'}(id) = B\ M_{B'B}(f)\ (B')^{-1} =$$

$$\begin{bmatrix} -1 & 5 \\ -2 & 9 \end{bmatrix} \begin{bmatrix} -2 & -6 & 2 & 0 \\ -3 & -9 & 3 & 0 \end{bmatrix} \begin{bmatrix} 19 & 3 & 2 & -22 \\ 23 & 3 & 3 & -27 \\ 19 & 4 & 1 & -22 \\ -7 & -2 & 0 & 8 \end{bmatrix}$$

$$= \begin{bmatrix} -897 & -104 & -130 & 1053 \\ -1587 & -184 & -230 & 1863 \end{bmatrix}.$$

Ejercicio 7. *Sea la aplicación* $f : \mathbb{R}^2 \to \mathbb{R}^4$ *tal que*

$$M_{C_2B_1}(f) = \begin{bmatrix} -2 & -8 \\ 2 & 9 \\ -1 & -4 \\ -1 & -1 \end{bmatrix}.$$

Calcula $M_{C_2B_2}(f)$ *siendo* B_1 *y* B_2 *las bases dadas por las matrices*

$$B_1 = \begin{bmatrix} 1 & 1 & -2 & -2 \\ 1 & 2 & -3 & -4 \\ 1 & -1 & 1 & 5 \\ -4 & -1 & 7 & 9 \end{bmatrix} \text{ y } B_2 = \begin{bmatrix} 1 & 2 & -1 & 8 \\ 0 & 1 & -1 & 0 \\ 0 & 5 & -4 & 5 \\ 0 & 5 & -6 & -4 \end{bmatrix}.$$

Solución:
Para pintar el diagrama puedes usar como base el siguiente dibujo:

$$\begin{tikzcd}
\mathbb{R}^2_C \arrow[r, "f", "M_1"'] \arrow[rr, bend right, "f", "M_3"'] & \mathbb{R}^4_{B_1} \arrow[r, "id", "M_2"'] & \mathbb{R}^4_{B_2}
\end{tikzcd}$$

Ejercicio 8. *Sea la aplicación $f : \mathbb{R}^3 \to \mathbb{R}^2$ tal que*

$$M_{B_1C}(f) = \begin{bmatrix} 4 & 0 & 8 \\ 3 & 0 & 6 \end{bmatrix}.$$

Calcula $M_{B_2C}(f)$ siendo B_1 y B_2 las bases dadas por las matrices

$$B_1 = \begin{bmatrix} -2 & 3 & -5 \\ 3 & -5 & 8 \\ 0 & -4 & 5 \end{bmatrix} \text{ y } B_2 = \begin{bmatrix} 1 & 1 & -2 \\ 3 & 4 & -8 \\ -5 & -3 & 7 \end{bmatrix}.$$

Solución:

```
MB11C2f = matrix(QQ,[[4,0,8],[3,0,6]])
B11=matrix(QQ,[[-2,3,-5],[3,-5,8],[0,-4,5]])
B12=matrix(QQ,[[1,1,-2],[3,4,-8],[-5,-3,7]])
```

Consideremos el siguiente diagrama:

$$\begin{tikzcd}
\mathbb{R}^3_{B_2} \arrow[r, "id", "M_{B_2B_1}(id)"'] \arrow[rr, bend right, "f", "M_{B_2C}(f)"'] & \mathbb{R}^3_{B_1} \arrow[r, "f", "M_{B_1C}(f)"'] & \mathbb{R}^2_C
\end{tikzcd}$$

Si calculamos cada una de las matrices de esas aplicaciones tenemos que

- $M_{B_2B_1}(id) = B_1^{-1}B_2$ porque B_1 es cuadrada y su inversa por la izquierda es su inversa.
- $M_{B_1C}(f)$ nos la dan.

Entonces

$$M_{B_2C}(f) = M_{B_1C}(f)\ M_{B_2B_1}(id) =$$

$$\begin{bmatrix} 4 & 0 & 8 \\ 3 & 0 & 6 \end{bmatrix} \begin{bmatrix} -2 & 3 & -5 \\ 3 & -5 & 8 \\ 0 & -4 & 5 \end{bmatrix}^{-1} \begin{bmatrix} 1 & 1 & -2 \\ 3 & 4 & -8 \\ -5 & -3 & 7 \end{bmatrix}$$

$$= \begin{bmatrix} -220 & -256 & 516 \\ -165 & -192 & 387 \end{bmatrix}.$$

Ejercicio 9. *Sea la aplicación $f : \mathbb{R}^2 \to \mathbb{R}^2$ tal que*

$$M_{B'B}(f) = \begin{bmatrix} -2 & 2 \\ -1 & 1 \end{bmatrix},$$

siendo B' y B las bases dadas por las matrices

$$B' = \begin{bmatrix} 6 & -5 \\ 5 & -4 \end{bmatrix} \text{ y } B = \begin{bmatrix} -1 & -5 \\ 2 & 9 \end{bmatrix}.$$

Calcula $M(f)$.

Solución:

```
MB1Bf = matrix(QQ,[[-2,2],[-1,1]])
B1=matrix(QQ,[[6,-5],[5,-4]])
B=matrix(QQ,[[-1,-5],[2,9]])
```

Consideremos el siguiente diagrama:

$$\begin{tikzcd}
\mathbb{R}^2_C \arrow[r, "id", "M_{CB'}(id)"'] \arrow[rrr, bend right, "f", "M_{CC}(f)"'] & \mathbb{R}^2_{B'} \arrow[r, "f", "M_{B'B}(f)"'] & \mathbb{R}^2_B \arrow[r, "id", "M_{BC}(id)"'] & \mathbb{R}^2_C
\end{tikzcd}$$

Si calculamos cada una de las matrices de esas aplicaciones tenemos que

- $M_{CB'}(id) = P_{CB'} = (B')^{-1}I = (B')^{-1}$ porque B' es cuadrada y su inversa por la izquierda es su inversa.
- $M_{B'B}(f)$ nos la dan.
- $M_{BC}(id) = P_{BC} = I^{-1}B = B$.

De donde

$$M(f) = M_{CC}(f) = M_{CC}(id \circ f \circ id) = M_{BC}(id)\ M_{B'B}(f)\ M_{CB'}(id) =$$

$$\begin{bmatrix} -1 & -5 \\ 2 & 9 \end{bmatrix} \begin{bmatrix} -2 & 2 \\ -1 & 1 \end{bmatrix} \begin{bmatrix} 6 & -5 \\ 5 & -4 \end{bmatrix}^{-1} = \begin{bmatrix} 7 & -7 \\ -13 & 13 \end{bmatrix}.$$

Ejercicio 10. *Sea la aplicación $f : \mathbb{Z}_5^2 \to \mathbb{Z}_5^3$ tal que*

$$M_{B'C}(f) = \begin{bmatrix} 2 & 3 \\ 4 & 4 \\ 1 & 3 \end{bmatrix},$$

Calcula $M_{C_2B}(f)$. siendo B' y B las bases dadas por las matrices

$$B' = \begin{bmatrix} 6 & -5 \\ 5 & -4 \end{bmatrix} \text{ y } B = \begin{bmatrix} 2 & 3 & 4 \\ 0 & -3 & 4 \\ 0 & 0 & 3 \end{bmatrix}.$$

Solución:

```
MB1C3f = matrix(Zmod(5),[[2,3],[4,4],[1,3]])
B1=matrix(Zmod(5),[[6,-5],[5,-4]])
B=matrix(Zmod(5),[[2,3,4],[0,-3,4],[0,0,3]])
```

Consideremos el siguiente diagrama:

$$\begin{tikzcd}
\mathbb{Z}^2_{5C} \arrow[r, "id", "M_{CB'}(id)"'] \arrow[rrr, bend right, "f", "M_{CB}(f)"'] & \mathbb{Z}^2_{5B'} \arrow[r, "f", "M_{B'C}(f)"'] & \mathbb{Z}^3_{5C} \arrow[r, "id", "M_{CB}(id)"'] & \mathbb{Z}^3_{5B}
\end{tikzcd}$$

Si calculamos cada una de las matrices de esas aplicaciones tenemos que

- $M_{CB'}(id) = P_{CB'} = (B')^{-1}I = (B')^{-1}$.

- $M_{B'C}(f)$ nos la dan.
- $M_{CB}(id) = P_{CB} = B^{-1}I = B^{-1}$.

De donde

$$M_{CB}(f) = M_{CB}(id \circ f \circ id) = M_{CB}(id)M_{B'C}(f)M_{CB'}(id)$$

$$= \begin{bmatrix} 2 & 3 & 4 \\ 0 & 2 & 4 \\ 0 & 0 & 3 \end{bmatrix}^{-1} \begin{bmatrix} 2 & 3 \\ 4 & 4 \\ 1 & 3 \end{bmatrix} \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}^{-1} = \begin{bmatrix} 0 & 2 \\ 3 & 0 \\ 2 & 1 \end{bmatrix}.$$

Ejercicio 11. *Calcula una base del espacio $N(A)$, anulador por la derecha de A siendo A la matriz*

$$\begin{bmatrix} 1 & 1 & 1 & 2 \\ 1 & 0 & 2 & 0 \\ 0 & 1 & 2 & 2 \end{bmatrix} \in \mathbf{M}_{3\times 4}(\mathbb{Z}_3)$$

Solución:

```
A=matrix(Zmod(3),[[1,1,1,2],[1,0,2,0],[0,1,2,2]])
R = A.echelon_form()
P = PolynomialRing(Zmod(3),4,'x')
X = matrix(P,4,1,P.gens())
Ecuaciones = A*X
EcuacionesR = R*X
v1 = column_matrix(Zmod(3),[1,1,1,0])
v2 = column_matrix(Zmod(3),[0,1,0,1])
```

$N(A)$ es el conjunto de soluciones del sistema

$$\begin{bmatrix} 1 & 1 & 1 & 2 \\ 1 & 0 & 2 & 0 \\ 0 & 1 & 2 & 2 \end{bmatrix} \begin{bmatrix} x_0 \\ x_1 \\ x_2 \\ x_3 \end{bmatrix} = 0$$

que escrito en forma de ecuaciones es

$$\begin{aligned} x_0 + x_1 + x_2 - x_3 &= 0 \\ x_0 - x_2 &= 0 \\ x_1 - x_2 - x_3 &= 0 \end{aligned}$$

Si reducimos por filas otenemos

$$\begin{bmatrix} 1 & 0 & 2 & 0 \\ 0 & 1 & 2 & 2 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

que corresponde al sistema de ecuaciones

$$\begin{aligned} x_0 - x_2 &= 0 \\ x_1 - x_2 - x_3 &= 0 \\ 0 &= 0 \end{aligned}$$

Si tomamos como parámetros las variables $x_2 = a$ y $x_3 = b$ podemos escribir las soluciones como

$$\begin{bmatrix} x_0 \\ x_1 \\ x_2 \\ x_3 \end{bmatrix} = a \begin{bmatrix} 1 \\ 1 \\ 1 \\ 0 \end{bmatrix} + b \begin{bmatrix} 0 \\ 1 \\ 0 \\ 1 \end{bmatrix}$$

Esto nos da la base que buscamos

$$\left\{ \begin{bmatrix} 1 \\ 1 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \\ 1 \end{bmatrix} \right\}$$

Ejercicio 12. *Calcula una base de $C(A)$, espacio generado por las columnas de A siendo A la matriz*

$$\begin{bmatrix} 4 & 4 & 2 \\ 2 & 1 & 0 \\ 2 & 2 & 2 \end{bmatrix} \in \mathbf{M}_{3\times 3}(\mathbb{Z}_5)$$

Solución:

```
A = matrix(Zmod(5),[[4,4,2],[2,1,0],[2,2,2]])
R = A.echelon_form()
```

Debemos encontrar el máximo número de columnas linealmente independientes. Reducimos la matriz A (bastaría con triangularizar) y obtenemos

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Todas las columnas son columnas pivote. Las columnas de la matriz original forman la base buscada:

$$\left\{ \begin{bmatrix} 4 \\ 2 \\ 2 \end{bmatrix}, \begin{bmatrix} 4 \\ 1 \\ 2 \end{bmatrix}, \begin{bmatrix} 2 \\ 0 \\ 2 \end{bmatrix} \right\}$$

Ejercicio 13. *Calcula una base de $C(A^T)$, espacio generado por las filas de A, siendo A la matriz*

$$\begin{bmatrix} 0 & 4 & 3 & 1 \\ 1 & 0 & 2 & 0 \\ 3 & 3 & 3 & 0 \end{bmatrix} \in \mathbf{M}_{3\times 4}(\mathbb{Z}_5)$$

Solución:

```
A=matrix(Zmod(5),[[0,4,3,1],[1,0,2,0],[3,3,3,0]])
B=A.echelon_form()
```

Para ello vamos a combinar las filas de A para conseguir vectores linealmente independientes. Obtenemos

$$\begin{bmatrix} 1 & 0 & 0 & 4 \\ 0 & 1 & 0 & 3 \\ 0 & 0 & 1 & 3 \end{bmatrix}$$

La base es:

$$\left\{ \begin{bmatrix} 1 \\ 0 \\ 0 \\ 4 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \\ 3 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ 3 \end{bmatrix} \right\}$$

Ejercicio 14. *Calcula una base de $N(A^T)$, espacio anulador por la izquierda de A, siendo A la matriz*

$$\begin{bmatrix} 0 & 2 \\ 0 & 1 \\ 2 & 0 \\ 2 & 1 \end{bmatrix} \in \mathbf{M}_{4\times 2}(\mathbb{Z}_5)$$

Solución:

```
A=matrix(Zmod(3),[[0,2],[0,1],[2,0],[2,1]])
B=block_matrix(1,2,[A,1])
R=B.echelon_form()
R=copy(R)
R.subdivide(2,2)
H = R.subdivision(1,1)
```

Para ello vamos y puesto que sólo nos interesa detectar las filas de ceros reducimos la matriz extendida con la matriz identidad:

$$\left[\begin{array}{cc|cccc} 0 & 2 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 1 & 0 & 0 \\ 2 & 0 & 0 & 0 & 1 & 0 \\ 2 & 1 & 0 & 0 & 0 & 1 \end{array}\right] \to \left[\begin{array}{cc|cccc} 1 & 0 & 0 & 0 & 2 & 0 \\ 0 & 1 & 0 & 0 & 2 & 1 \\ \hline 0 & 0 & 1 & 0 & 2 & 1 \\ 0 & 0 & 0 & 1 & 1 & 2 \end{array}\right].$$

Si nos quedamos con las filas a la derecha de las filas nulas de la reducción, la base es:

$$\left\{ \begin{bmatrix} 1 \\ 0 \\ 2 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 1 \\ 2 \end{bmatrix} \right\}$$

Ejercicio 15. *Sea $V = C(B)$ el espacio generado por las columnas de la matriz*

$$B = \begin{bmatrix} 3 & 4 & 4 & 0 \\ 1 & 0 & 0 & 0 \\ 4 & 2 & 3 & 3 \end{bmatrix} \in \mathbf{M}_{3\times 4}(\mathbb{Z}_5)$$

Extrae una base de entre los vectores columna de B.

Solución:

```
B=matrix(Zmod(5),[[3,4,4,0],[1,0,0,0],[4,2,3,3]])
C=B.echelon_form()
```

Para encontrar la base debemos encontrar el máximo número de columnas libres es decir pivote. Reducimos B (bastaría con triangularizar) y obtenemos

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 2 \\ 0 & 0 & 1 & 3 \end{bmatrix}$$

Las columnas pivote son 1, 2 y 3, por lo tanto la base es

$$\left\{ \begin{bmatrix} 3 \\ 1 \\ 4 \end{bmatrix}, \begin{bmatrix} 4 \\ 0 \\ 2 \end{bmatrix}, \begin{bmatrix} 4 \\ 0 \\ 3 \end{bmatrix} \right\}$$

Ejercicio 16. *Sea V un espacio vectorial con base B y consideremos los vectores linealmente independientes de V dados por las columnas de B'. Extiende B' usando columnas de B para formar una base de V, siendo*

$$B = \begin{bmatrix} 1 & -4 \\ -1 & 5 \\ -3 & 7 \\ 1 & -2 \end{bmatrix} \in \mathbf{M}_{4\times 2}(\mathbb{R}) \qquad B' = \begin{bmatrix} 5 \\ -6 \\ -10 \\ 3 \end{bmatrix} \in \mathbf{M}_{4\times 1}(\mathbb{R})$$

Solución:

```
B=matrix(QQ,[[1,-4],[-1,5],[-3,7],[1,-2]])
B1=column_matrix(QQ,[5,-6,-10,3])
B1B=block_matrix(1,2,[B1,B])
C=B1B.echelon_form()
```

La base que nos piden la obtendremos a través del conjunto generador de V dado por los vectores $[B'|B]$. Al reducirla (bastaría con triangularizar) obtenemos

$$\left[\begin{array}{c|cc} 1 & 0 & -1 \\ 0 & 1 & 1 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{array}\right].$$

Las columnas pivote son 1 y 2, por lo que esas son precisamente las columnas de $[B'|B]$ que son base buscada:

$$\left\{ \begin{bmatrix} 5 \\ -6 \\ -10 \\ 3 \end{bmatrix}, \begin{bmatrix} 1 \\ -1 \\ -3 \\ 1 \end{bmatrix} \right\}$$

Ejercicio 17. *Sea $V = C(B)$ y $U = C(B')$. Determina si el espacio U es un subespacio de V, siendo*

$$B = \begin{bmatrix} 0 & 3 & 2 & 4 \\ 3 & 1 & 0 & 1 \\ 3 & 0 & 2 & 4 \\ 0 & 0 & 3 & 3 \\ 2 & 4 & 2 & 1 \end{bmatrix} \in \mathbf{M}_{5\times 4}(\mathbb{Z}_5) \qquad B' = \begin{bmatrix} 2 & 4 \\ 2 & 4 \\ 0 & 2 \\ 1 & 3 \\ 2 & 3 \end{bmatrix} \in \mathbf{M}_{5\times 2}(\mathbb{Z}_5)$$

Solución: Si $V = C(B)$ y $U = C(B')$, entonces $U \subseteq V$ si y sólo si todas las columnas de B' se pueden expresar como combinación lineal de las columnas de B, lo que equivale a que al reducir por filas la matriz $[B|B']$, las columnas de B' no son columnas pivote.

```
B=matrix(Zmod(5),[[0,3,2,4],[3,1,0,1],
[3,0,2,4],[0,0,3,3],[2,4,2,1]])
B1=matrix(Zmod(5),[[2,4],[2,4],
[0,2],[1,3],[2,3]])
BB1 = block_matrix(1,2,[B,B1])
R=BB1.echelon_form()
```

Ampliamos la matriz B a su derecha con la matriz B'.

$$[B|B'] = \left[\begin{array}{cccc|cc} 0 & 3 & 2 & 4 & 2 & 4 \\ 3 & 1 & 0 & 1 & 2 & 4 \\ 3 & 0 & 2 & 4 & 0 & 2 \\ 0 & 0 & 3 & 3 & 1 & 3 \\ 2 & 4 & 2 & 1 & 2 & 3 \end{array}\right]$$

Reduciendo por filas obtenemos la matriz (sobraría con triangularizar)

$$\left[\begin{array}{cccc|cc} 1 & 0 & 0 & 4 & 2 & 0 \\ 0 & 1 & 0 & 4 & 1 & 4 \\ 0 & 0 & 1 & 1 & 2 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{array}\right]$$

Vemos que ninguna columna de la matriz B' es pivote, por lo tanto todos los vectores de B' están en V y $U \leq V$.

Ejercicio 18. *Sea $V = C(B)$ y $U = C(B')$. Determina si el espacio U es un subespacio de V, siendo*

$$B = \begin{bmatrix} 3 & 0 & 4 & 1 & 3 \\ 6 & 6 & 4 & 4 & 2 \\ 4 & 0 & 5 & 0 & 0 \\ 6 & 3 & 1 & 0 & 1 \\ 4 & 5 & 4 & 0 & 0 \\ 1 & 6 & 3 & 0 & 3 \end{bmatrix} \in \mathbf{M}_{6\times 5}(\mathbb{Z}_7) \qquad B' = \begin{bmatrix} 2 & 3 & 2 & 3 & 6 \\ 6 & 6 & 1 & 3 & 1 \\ 0 & 2 & 6 & 0 & 5 \\ 4 & 5 & 4 & 3 & 5 \\ 2 & 2 & 1 & 5 & 6 \\ 3 & 0 & 3 & 0 & 5 \end{bmatrix} \in \mathbf{M}_{6\times 5}(\mathbb{Z}_7)$$

Solución:

Ejercicio 19. *Sea $V = N(H)$ y $U = N(H')$. Determina si el espacio V es un subespacio de U, siendo*

$$H = \begin{bmatrix} 1 & 4 & 4 & 2 & 0 \\ 0 & 2 & 4 & 2 & 4 \\ 2 & 1 & 4 & 2 & 1 \\ 2 & 2 & 1 & 3 & 3 \end{bmatrix} \in \mathbf{M}_{4\times 5}(\mathbb{Z}_5) \qquad H' = \begin{bmatrix} 4 & 2 & 3 & 4 & 2 \\ 1 & 1 & 3 & 4 & 4 \end{bmatrix} \in \mathbf{M}_{2\times 5}(\mathbb{Z}_5)$$

Solución:

```
H=matrix(Zmod(5),[[1,4,4,2,0],[0,2,4,2,4],
[2,1,4,2,1],[2,2,1,3,3]])
H1=matrix(Zmod(5),[[4,2,3,4,2],[1,1,3,4,4]])
HTI = block_matrix(1,2,[H.T,1])
R=HTI.echelon_form()
R=copy(R)
R.subdivide(2,4)
A=R.subdivision(1,1)
```

Para ver si V es un subespacio de U, pasamos V a paramétricas, $V = C(A)$. Entonces como $V = C(A)$ y $U = N(H')$, $V \leq U$ si y sólo si $H' \cdot A = 0$.

Determinamos la matriz $[H^T|I]$

$$[H^T|I] = \left[\begin{array}{cccc|ccccc} 1 & 0 & 2 & 2 & 1 & 0 & 0 & 0 & 0 \\ 4 & 2 & 1 & 2 & 0 & 1 & 0 & 0 & 0 \\ 4 & 4 & 4 & 1 & 0 & 0 & 1 & 0 & 0 \\ 2 & 2 & 2 & 3 & 0 & 0 & 0 & 1 & 0 \\ 0 & 4 & 1 & 3 & 0 & 0 & 0 & 0 & 1 \end{array}\right]$$

Reducimos por filas

$$\left[\begin{array}{cccc|ccccc} 1 & 0 & 2 & 2 & 0 & 0 & 0 & 3 & 1 \\ 0 & 1 & 4 & 2 & 0 & 0 & 0 & 0 & 4 \\ \hline 0 & 0 & 0 & 0 & 1 & 0 & 0 & 2 & 4 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 3 & 3 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 3 & 0 \end{array}\right]$$

Las paramétricas de V son $V = C(A)$ donde A es la matriz

$$A = \left[\begin{array}{ccc} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \\ 2 & 3 & 3 \\ 4 & 3 & 0 \end{array}\right]$$

$$H' \cdot A = \left[\begin{array}{ccccc} 4 & 2 & 3 & 4 & 2 \\ 1 & 1 & 3 & 4 & 4 \end{array}\right] \cdot \left[\begin{array}{ccccc} 1 & 0 & 0 & 2 & 4 \\ 0 & 1 & 0 & 3 & 3 \\ 0 & 0 & 1 & 3 & 0 \end{array}\right] = \left[\begin{array}{ccc} 0 & 0 & 0 \\ 0 & 0 & 0 \end{array}\right]$$

Como el resultado es 0, concluimos que $V \leq U$.

Ejercicio 20. *Sea $V = N(H)$ y $U = N(H')$. Determina si el espacio V es un subespacio de U, siendo*

$$H = \left[\begin{array}{ccccc} 12 & 19 & 9 & 30 & 0 \\ 30 & 14 & 29 & 4 & 19 \\ 28 & 14 & 11 & 29 & 28 \\ 3 & 1 & 12 & 25 & 13 \end{array}\right] \in \mathbf{M}_{4\times 5}(\mathbb{Z}_{31}) \qquad H' = \left[\begin{array}{ccccc} 26 & 24 & 4 & 20 & 0 \\ 10 & 6 & 15 & 7 & 4 \\ 22 & 0 & 2 & 25 & 26 \end{array}\right] \in \mathbf{M}_{3\times 5}(\mathbb{Z}_{31})$$

Solución:

```
H=matrix(Zmod(31),[[12,19,9,30,0],[30,14,29,4,19],
[28,14,11,29,28],[3,1,12,25,13]])
H1=matrix(Zmod(31),[[26,24,4,20,0],[10,6,15,7,4],
[22,0,2,25,26]])
HTI = block_matrix(1,2,[H.T,1])
R=HTI.echelon_form()
R=copy(R)
R.subdivide(2,4)
A=R.subdivision(1,1)
```

Para ver si V es un subespacio de U, pasamos V a paramétricas, $V = C(A)$. Entonces como $V = C(A)$ y $U = N(H')$, $V \leq U$ si y sólo si $H' \cdot A = 0$.

Determinamos la matriz $[H^T|I]$

$$[H^T|I] = \left[\begin{array}{rrrr|rrrrr} 12 & 30 & 28 & 3 & 1 & 0 & 0 & 0 & 0 \\ 19 & 14 & 14 & 1 & 0 & 1 & 0 & 0 & 0 \\ 9 & 29 & 11 & 12 & 0 & 0 & 1 & 0 & 0 \\ 30 & 4 & 29 & 25 & 0 & 0 & 0 & 1 & 0 \\ 0 & 19 & 28 & 13 & 0 & 0 & 0 & 0 & 1 \end{array}\right]$$

Reducimos por filas

$$\left[\begin{array}{rrrr|rrrrr} 1 & 0 & 3 & 12 & 0 & 0 & 0 & 30 & 10 \\ 0 & 1 & 8 & 17 & 0 & 0 & 0 & 0 & 18 \\ \hline 0 & 0 & 0 & 0 & 1 & 0 & 0 & 12 & 22 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 19 & 23 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 9 & 8 \end{array}\right]$$

Las paramétricas de V son $V = C(A)$ donde A es la matriz

$$A = \left[\begin{array}{rrr} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \\ 12 & 19 & 9 \\ 22 & 23 & 8 \end{array}\right]$$

$$H' \cdot A = \left[\begin{array}{rrrrr} 26 & 24 & 4 & 20 & 0 \\ 10 & 6 & 15 & 7 & 4 \\ 22 & 0 & 2 & 25 & 26 \end{array}\right] \cdot \left[\begin{array}{rrrrr} 1 & 0 & 0 & 12 & 22 \\ 0 & 1 & 0 & 19 & 23 \\ 0 & 0 & 1 & 9 & 8 \end{array}\right] = \left[\begin{array}{rrr} 18 & 1 & 29 \\ 27 & 14 & 17 \\ 26 & 19 & 1 \end{array}\right]$$

Dicho producto no es la matriz nula, por tanto V no está contenido en U.

Ejercicio 21. *Sea $V = C(B)$ y $U = N(H')$. Determina el espacio $U + V$ en paramétricas, siendo*

$$B = \left[\begin{array}{r} 1 \\ 8 \\ 14 \\ 0 \\ 18 \end{array}\right] \in \mathbf{M}_{5\times 1}(\mathbb{Z}_{19}) \qquad H' = \left[\begin{array}{rrrrr} 17 & 6 & 15 & 5 & 3 \\ 13 & 6 & 12 & 4 & 0 \end{array}\right] \in \mathbf{M}_{2\times 5}(\mathbb{Z}_{19})$$

Solución:

```
B=column_matrix(Zmod(19),[1,8,14,0,18])
H1=matrix(Zmod(19),[[17,6,15,5,3],[13,6,12,4,0]])
H1TI = block_matrix(1,2,[H1.T,1])
R=H1TI.echelon_form()
R=copy(R)
R.subdivide(2,2)
A=R.subdivision(1,1)
D=block_matrix(1,2,[A.T,B])
```

Para determinar el espacio $U + V$ en paramétricas, debemos en primer lugar determinar las paramétricas del espacio U,¡para ello reducimos la matriz $[H'^T|I]$.

$$[H'^T|I] = \left[\begin{array}{rr|rrrrr} 17 & 13 & 1 & 0 & 0 & 0 & 0 \\ 6 & 6 & 0 & 1 & 0 & 0 & 0 \\ 15 & 12 & 0 & 0 & 1 & 0 & 0 \\ 5 & 4 & 0 & 0 & 0 & 1 & 0 \\ 3 & 0 & 0 & 0 & 0 & 0 & 1 \end{array}\right]$$

Reducimos por filas

$$\left[\begin{array}{rr|rrrrr} 1 & 0 & 0 & 0 & 0 & 0 & 13 \\ 0 & 1 & 0 & 0 & 0 & 5 & 17 \\ \hline 0 & 0 & 1 & 0 & 0 & 11 & 14 \\ 0 & 0 & 0 & 1 & 0 & 8 & 10 \\ 0 & 0 & 0 & 0 & 1 & 16 & 0 \end{array}\right]$$

Las paramétricas de U son $U = C(A)$ donde A es la matriz

$$A = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \\ 11 & 8 & 16 \\ 14 & 10 & 0 \end{bmatrix}$$

Unas ecuaciones paramétricas de $U+V$ son

$$U + V = C \left[\begin{array}{rrr|r} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & 8 \\ 0 & 0 & 1 & 14 \\ 11 & 8 & 16 & 0 \\ 14 & 10 & 0 & 18 \end{array}\right]$$

Ejercicio 22. *Sea $V = C(B)$ y $U = N(H')$. Determina el espacio $U \cap V$ en implícitas, siendo*

$$B = \begin{bmatrix} 1 \\ 8 \\ 15 \\ 2 \\ 16 \end{bmatrix} \in \mathbf{M}_{5\times 1}(\mathbb{Z}_{19}) \qquad H' = \begin{bmatrix} 1 & 11 & 0 & 10 & 0 \\ 5 & 15 & 6 & 2 & 4 \end{bmatrix} \in \mathbf{M}_{2\times 5}(\mathbb{Z}_{19})$$

Solución: