AMD CURSO 2023-2024. PRÁCTICAS DE LA SEMANA 13. GEOMETRÍA AFÍN

Cuestiones teóricas previas

Una aplicación $f : \mathcal{A}^n(\mathbb{R}) \to \mathcal{A}^n(\mathbb{R})$ se dice **afín** si existen una matriz A de orden $n \times n$ y un punto $Q \in \mathbb{R}^n$ tales que $f(P) = AP + Q$, lo cual se expresa con matrices del siguiente modo:

$$\left[\frac{1}{f(P)}\right] = \left[\frac{1 \mid 0}{Q \mid A}\right]\left[\frac{1}{Q}\right]$$

Entre las aplicaciones afines (en el plano y en el espacio), nos interesan especialmente un grupo de transformaciones geométricas que reciben el nombre de **transformaciones afines**, y son las siguientes:

- Transformaciones afines en el plano:
  - Giro alrededor de un punto.
  - Proyección ortogonal sobre una recta
  - Simetría ortogonal sobre una recta
  - Homotecia
- Transformaciones afines en el espacio:
  - Giro alrededor de un recta.
  - Proyección ortogonal sobre una recta o un plano
  - Simetría ortogonal sobre una recta o un plano
  - Homotecia

La matriz M_T asociada a cada una de estas transformaciones T tiene una forma especialmente sencilla (su expresión canónica) porque se toma respecto de un sistema de referencia afín R que se adapta especialmente bien a lo que hace la transformación afín T. Por ejemplo, si $T = G_\alpha$ es un giro en el plano de ángulo α, centrado en el punto Q, entonces

$$R = \left[\frac{1 \mid 0}{Q \mid C}\right],$$

donde C es la base canónica de $\mathbb{R}^2$, y

$$M_{G_\alpha} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos\alpha & -\sin\alpha \\ 0 & \sin\alpha & \cos\alpha \end{bmatrix}$$

Entonces la matriz asociada, $M(T)$, respecto del sistema de referencia canónico del espacio afín, se expresa en términos de M_T como:

$$M(T) = RM_TR^{-1}$$

En lo sucesivo, denotaremos como $M_{RR}(T)$ a la matriz M_T para dejar constancia, en cada caso, de cuál es el sistema de referencia afín especial, R, que se ha usado para obtener la expresión canónica de la transformación T -y que luego usamos para obtener la matriz $M(T)$, que es la asociada a T respecto del sistema de referencia canónico de $\mathcal{A}^n(\mathbb{R})$.

Ejercicio 1.

a) *Calcula la matriz de la aplicación afín* $g : \mathcal{A}^2(\mathbb{R}) \to \mathcal{A}^2(\mathbb{R})$ *correspondiente al* GIRO *de* $30°$ *alrededor del punto* $\begin{pmatrix} 2 \\ -1 \end{pmatrix}$.

b) *Sea* $C \subseteq \mathcal{A}^2(\mathbb{R})$ *el hexágono regular centrado en el punto* $\begin{pmatrix} 4 \\ 4 \end{pmatrix}$ *tal que uno de sus vértices es el punto* $\begin{pmatrix} 5 \\ 4 \end{pmatrix}$. *Calcula todos los vértices de* C *y* $g(C)$. *Dibuja los hexágonos* C, $g(C)$ *y las líneas que unen los centros de los hexágonos con el centro de giro.*

Solución:
a.

```
R = matrix(RR, [[1, 0, 0], [2, 1, 0], [-1, 0, 1]])
alpha = pi/6
MRRg = matrix(RR, [[1, 0, 0], [0, cos(alpha), -sin(alpha)], [0, sin(alpha), cos(alpha)]])

Mg = R * MRRg * R^-1
```

$$R = \begin{bmatrix} 1 & 0 & 0 \\ 2 & 1 & 0 \\ -1 & 0 & 1 \end{bmatrix}$$

$$M_{RR}(g) = \begin{bmatrix} 1 & 0 & 0 \\ 0 & cos(30) & -sin(30) \\ 0 & sin(30) & cos(30) \end{bmatrix}$$

$$M = RM_{RR}(g)R^{-1} = \begin{pmatrix} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ -0.232050807568877 & 0.866025403784439 & -0.500000000000000 \\ -1.13397459621556 & 0.500000000000000 & 0.866025403784439 \end{pmatrix}$$

b.

```
R1 = matrix(RR, [[1, 0, 0], [4, 1, 0], [4, 0, 1]])

b = pi/3
MRRg = matrix(RR, [[1, 0, 0], [0, cos(b), -sin(b)], [0, sin(b), cos(b)]])

Mg1 = R1 * MRRg * R1^-1

C = [Mg1^i * vector(RR, [1, 5, 4]) for i in range(6)]
gC = [Mg * v for v in C]
gCo = (Mg * vector(RR, [1, 4, 4]))[1:]

C = [v[1:] for v in C]
gC = [v[1:] for v in gC]
```



Ejercicio 2. *Sea $r \subseteq \mathcal{A}^2(\mathbb{R})$ la recta que pasa por los puntos $\begin{pmatrix} -1 \\ 0 \end{pmatrix}$ y $\begin{pmatrix} 2 \\ 2 \end{pmatrix}$.*

a) *Calcula las matrices de las aplicaciones afines $p, s : \mathcal{A}^2(\mathbb{R}) \to \mathcal{A}^2(\mathbb{R})$ correspondientes a la* PROYECCIÓN *y a la* SIMETRÍA *respecto de r.*

b) *Sea $C \subseteq \mathcal{A}^2(\mathbb{R})$ el cuadrado centrado en el punto $\begin{pmatrix} 3 \\ -3 \end{pmatrix}$ y con uno de sus vértices en $\begin{pmatrix} 4 \\ -2 \end{pmatrix}$.*
- *Calcula todos los vértices de C y $s(C)$.*
- *Dibuja r, C, $s(C)$, $p(C)$ y las líneas que unen los vértices del cuadrado C con su simétrico $s(C)$.*

Solución:

```
pr1 = vector(RR, [-1, 0])
pr2 = vector(RR, [2, 2])
vr = (pr2 - pr1)
u1 = vr.normalized()
u2 = vector(RR, [vr[1], -vr[0]]).normalized()

R = matrix(RR, [[1, 0, 0], [pr1[0], u1[0], u2[0]], [pr1[1], u1[1], u2[1]]])
MRRp = matrix(RR, [[1, 0, 0], [0, 1, 0], [0, 0, 0]])
MRRs = matrix(RR, [[1, 0, 0], [0, 1, 0], [0, 0, -1]])

Mp = R * MRRp * R^-1
Ms = R * MRRs * R^-1
```

a.

$$R = \begin{pmatrix} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ -1.00000000000000 & 0.832050294337844 & 0.554700196225229 \\ 0.000000000000000 & 0.554700196225229 & -0.832050294337844 \end{pmatrix}$$

$$M_p = \begin{pmatrix} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ -0.307692307692308 & 0.692307692307692 & 0.461538461538462 \\ 0.461538461538462 & 0.461538461538462 & 0.307692307692308 \end{pmatrix}$$

$$M_s = \begin{pmatrix} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ -0.615384615384615 & 0.384615384615385 & 0.923076923076923 \\ 0.923076923076923 & 0.923076923076923 & -0.384615384615385 \end{pmatrix}$$

b.

```
R = matrix(RR, [[1, 0, 0], [3, 1, 0], [-3, 0, 1]])

a = pi/2
MRRg = matrix(RR, [[1, 0, 0], [0, cos(a), -sin(a)], [0, sin(a), cos(a)]])
Mg = R * MRRg * R^-1

cp0 = vector(RR, [1, 4, -2])

C = [Mg^i * cp0 for i in range(4)]
sC = [Ms * v for v in C]
pC = [Mp * v for v in C]

C = [v[1:] for v in C]
sC = [v[1:] for v in sC]
pC = [v[1:] for v in pC]
```



Ejercicio 3.

a) *Calcula la matriz de la aplicación afín* $h : \mathcal{A}^2(\mathbb{R}) \to \mathcal{A}^2(\mathbb{R})$ *correspondiente a la* HOMOTECIA *con factor* 3 *y con centro en el punto* $P = \begin{pmatrix} 1 \\ 1 \end{pmatrix}$.

b) *Sea* $C \subseteq \mathcal{A}^2(\mathbb{R})$ *el triángulo isósceles contenido en el segundo cuadrante que tiene altura* 5 *y cuya base es el segmento con extremos* $\begin{pmatrix} -4 \\ 2 \end{pmatrix}$ *y* $\begin{pmatrix} -3 \\ 2 \end{pmatrix}$.
   - *Calcula todos los vértices de* C *y* $h(C)$.
   - *Dibuja* P, C *y* $h(C)$.

Solución:

a.

```
R = matrix(RR, [[1, 0, 0], [1, 1, 0], [1, 0, 1]])
f = 3
MRRh = matrix(RR, [[1, 0, 0], [0, f, 0], [0, 0, f]])

Mf = R * MRRh * R^-1

tp0 = vector(RR, [-4, 2])
tp1 = vector(RR, [-3, 2])
tsv = tp1 - tp0
tsmp = tp0 + tsv/2
tp2 = tsmp + 5 * vector(RR, [-tsv[1], tsv[0]]).normalized()

C = vector(RR, [1, tp0[0], tp0[1]]), vector(RR, [1, tp1[0], tp1[1]]), vector(RR, [1, tp2[0], tp2[1]])
hC = [Mf * v for v in C]

C = [v[1:] for v in C]
hC = [v[1:] for v in hC]
```



Ejercicio 4. *Calcula los vértices de un octógono regular en* $\mathcal{A}^2(\mathbb{R})$ *centrado en el punto* $\begin{bmatrix} 1 \\ 3 \end{bmatrix}$*, sabiendo que el punto* $\begin{bmatrix} 2 \\ 5 \end{bmatrix}$ *es un vértice de dicho octógono. Pinta el octógono en color amarillo y sus vértices en color azul.*

Solución:

```
R = matrix(RR, [[1, 0, 0], [1, 1, 0], [3, 0, 1]])

a = pi/4
MRRg = matrix(RR, [[1, 0, 0], [0, cos(a), -sin(a)], [0, sin(a), cos(a)]])
Mg = R * MRRg * R^-1

op0 = vector(RR, [1, 2, 5])

C = [Mg^i * op0 for i in range(8)]
C = [v[1:] for v in C]
```



Ejercicio 5. *Se dibuja una estrella regular de 5 puntas con centro en el punto* $O_1 = \begin{bmatrix} -4 \\ -1 \end{bmatrix}$ *y con vértice* $A = \begin{bmatrix} -2 \\ 0 \end{bmatrix}$. *El resto de los vértices se nombran* B, C, D *y* E *tal y como indica la figura. Posteriormente, se gira la estrella un ángulo de* $48°$ *alrededor del punto* $P = \begin{bmatrix} -8 \\ 4 \end{bmatrix}$ *en sentido positivo.*



Determinar las coordenadas x e y de los vértices $A, B, C, D, E, A', B', C', D', E'$ *y dibuja las estrellas y las líneas que unen los centros de dichas estrellas con el centro de giro.*

Solución:

```
R = matrix(RR, [[1, 0, 0], [-4, 1, 0], [-1, 0, 1]])

a = 2*pi/5
MRRg = matrix(RR, [[1, 0, 0], [0, cos(a), -sin(a)], [0, sin(a), cos(a)]])
Mg = R * MRRg * R^-1

sp0 = vector(RR, [1, -2, 0])

C = [Mg^i * sp0 for i in range(5)]

R = matrix(RR, [[1, 0, 0], [-8, 1, 0], [4, 0, 1]])

a = pi/180 * 48
MRRg = matrix(RR, [[1, 0, 0], [0, cos(a), -sin(a)], [0, sin(a), cos(a)]])
Mg = R * MRRg * R^-1

Cp = [Mg * v for v in C]

C = [v[1:] for v in C]
Cp = [v[1:] for v in Cp]
```


O

Cuestiones teóricas previas

La recta en el plano que pasa por los puntos de coordenadas $P(a,b)$ y $Q(c,d)$ se puede calcular de la forma

$$\begin{vmatrix} 1 & 1 & 1 \\ x & a & c \\ y & b & d \end{vmatrix} = 0$$

La distancia de un punto $P = (x_0, x_1)$ sobre una recta r de ecuación $Ax + By + C = 0$ es

$$d(P,r) = \frac{|Ax_0 + Bx_1 + C|}{\sqrt{A^2 + B^2}}.$$

Ejercicio 6. *Determina una estrella regular de 5 puntas con centro en el punto $O_1 = (5,0)$ y con vértice $A = (6,0)$. El resto de los vértices se nombran B, C, D y E tal y como indica la figura. Posteriormente gira la estrella un ángulo de $39°$ alrededor del punto $P = (1,2)$ en sentido positivo.*



Dibujar dicha estrella así como su girada, mostrando además los segmentos PO_1 y PO_1'. Además calcular los siguientes puntos y valores:

a) *La ecuación de la recta PO_1.*
b) *La distancia de A a la recta PO_1*
c) *Las coordenadas del punto B.*
d) *Las coordenadas del punto O_1' centro de la estrella girada.*
e) *Las coordenadas del punto B', vértice correspondiente con el vértice B de la estrella girada.*

Solución:

Introducimos el sistema de referencia canónico con punto (5,0) y base e_1, e_2, R:

```
R = matrix(RR,[[1,0,0],
              [5,1,0],
              [0,0,1]])
R.subdivide(1,1)
```

$$R = \left(\begin{array}{c|cc} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline 5.00000000000000 & 1.00000000000000 & 0.000000000000000 \\ 0.000000000000000 & 0.000000000000000 & 1.00000000000000 \end{array}\right)$$

El sistema de referencia para realizar el giro y obtener las estrella de vértices A, B, C, D y E será g. Para obtener los vértices de la estrella, definimos el ángulo $\beta = 2\pi/5$ y la matriz del giro de ángulo β en este sistema de referencia

```
beta = 2*pi/5
G = matrix(RR,[[1, 0        ,  0        ],
               [0, cos(beta), -sin(beta)],
               [0, sin(beta),  cos(beta)]])
g=R*G*R^-1
g.subdivide(1,1)
```

La matriz del giro del sistema de referencia que resulta es

$$g = RG_{\frac{2\pi}{5}}R^{-1} = \left(\begin{array}{r|rr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline 3.45491502812526 & 0.309016994374947 & -0.951056516295154 \\ -4.75528258147577 & 0.951056516295154 & 0.309016994374947 \end{array}\right).$$

Definimos los vértices del pentágono girando el vértice $A = \begin{pmatrix} 6 \\ 0 \end{pmatrix}$ en coordenadas homogéneas

```
V = [g^i*vector(RR,[1,6,0]) for i in range(5)]
```

Los vértices con coordenada homogénea son:

$$V[0] = (1,6,0) = (1.00000000000000,\ 6.00000000000000,\ 0.000000000000000)$$
$$V[1] = g(1,6,0) = (1.00000000000000,\ 5.30901699437495,\ 0.951056516295154)$$
$$V[2] = g^2(1,6,0) = (1.00000000000000,\ 4.19098300562505,\ 0.587785252292474)$$
$$V[3] = g^3(1,6,0) = (1.00000000000000,\ 4.19098300562505,\ -0.587785252292473)$$
$$V[4] = g^4(1,6,0) = (1.00000000000000,\ 5.30901699437495,\ -0.951056516295154)$$

Ahora debemos girar dichos vértices un ángulo de 39° centrado en el punto $P = (1, 2)$. Para eso introducimos como R_1 el sistema de referencia canónico en P.

```
R1 = matrix(RR,[[1,0,0],
                [1,1,0],
                [2,0,1]])
R1.subdivide(1,1)
```

Nuestro sistema de referencia ahora es

$$R_1 = \left(\begin{array}{r|rr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline 1.00000000000000 & 1.00000000000000 & 0.000000000000000 \\ 2.00000000000000 & 0.000000000000000 & 1.00000000000000 \end{array}\right)$$

Definimos el ángulo $\alpha = 39\pi/180$ y la matriz del giro de ángulo α en este sistema de referencia

```
alpha = 39*pi/180
G1 = matrix(RR,[[1, 0         ,  0         ],
                [0, cos(alpha), -sin(alpha)],
                [0, sin(alpha),  cos(alpha)]])
g1=R1*G1*R1^-1
g1.subdivide(1,1)
```

La matriz del giro que resulta es

$$g_1 = R_1G_{\frac{39\pi}{180}}R_1^{-1} = \left(\begin{array}{r|rr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline 1.48149482064270 & 0.777145961456971 & -0.629320391049838 \\ -0.183612313963779 & 0.629320391049838 & 0.777145961456971 \end{array}\right).$$

Los vértices en coordenadas homogéneas de la estrella girada con centro en el punto P son

```
V1 = [g1*v for v in V]
```

$$V_1[0] = g_1(1,6,0) = (1.00000000000000,\ 6.14437058938453,\ 3.59231003233525)$$
$$V_1[1] = g_1(g(1,6,0)) = (1.00000000000000,\ 5.00885667838226,\ 3.89657006778261)$$
$$V_1[2] = g_1(g^2(1,6,0)) = (1.00000000000000,\ 4.36859509317298,\ 2.91065368504247)$$
$$V_1[3] = g_1(g^3(1,6,0)) = (1.00000000000000,\ 5.10840558282504,\ 1.99706381499634)$$
$$V_1[4] = g_1(g^4(1,6,0)) = (1.00000000000000,\ 6.20589519587298,\ 2.41835060627038)$$

Quitamos la coordenada homogénea para poder representar los puntos en el plano

```
V = [v[1:] for v in V]
V1 = [v[1:] for v in V1]
```

De esta forma los puntos ahora son Dibujamos las estrellas así como los puntos P, $O1$, $O'1$ y los segmentos PO_1 y PO_1'.

$$V[0] = (6.00000000000000, 0.000000000000000) \quad V_1[0] = (6.14437058938453, 3.59231003233525)$$
$$V[1] = (5.30901699437495, 0.951056516295154) \quad V_1[1] = (5.00885667838226, 3.89657006778261)$$
$$V[2] = (4.19098300562505, 0.587785252292474) \quad V_1[2] = (4.36859509317298, 2.91065368504247)$$
$$V[3] = (4.19098300562505, -0.587785252292473) \quad V_1[3] = (5.10840558282504, 1.99706381499634)$$
$$V[4] = (5.30901699437495, -0.951056516295154) \quad V_1[4] = (6.20589519587298, 2.41835060627038)$$

Para introducir los comandos en Sagesub adecuadamente debemos introducir los siguientes puntos y vectores:

```
P=vector(RR,[1,2])
O1=vector(RR,[5,0])
g1O1=g1*vector(RR,[1,5,0])
g1O1=g1O1[1:]
```

$$P = (1.00000000000000, 2.00000000000000)$$
$$O_1 = (5.00000000000000, 0.000000000000000)$$
$$g_1(O_1) = (5.36722462792756, 2.96298964128541)$$

Introduciendo los siguientes comandos:

```
\begin{sagesub}
\begin{center}
\begin{tikzpicture}[scale = 1.3,
cara1/.style={thick, color = blue, fill opacity = 0.3, fill = blue!20},
cara2/.style={thick, color = red, fill opacity = 0.3, fill = red!20}]
\draw[->,thick,gray] (-1,0) -- (7,0); % Eje X
\draw[->,thick,gray] (0,-1) -- (0,5); % Eje Y
\draw[cara1] !{V[0]} -- !{V[2]} -- !{V[4]} -- !{V[1]} -- !{V[3]} -- cycle;
\draw[cara2] !{V1[0]} -- !{V1[2]} -- !{V1[4]} -- !{V1[1]} -- !{V1[3]} -- cycle;
\draw[dashed] !{P} -- !{O1};
\draw[dashed] !{P} -- !{g1O1};
\draw[fill = white] !{P} circle (0.5mm);
\draw[fill = white] !{P} circle (0.5mm) node[above left] {$P$};
\draw[fill = white] !{O1} circle (0.5mm);
\draw[fill = white] !{g1O1} circle (0.5mm);
\end{tikzpicture}
\end{center}
\end{sagesub}
```

se obtiene:



Además

a) Ecuación de la recta PO_1.

```
Pol.<x,y> = PolynomialRing(RR)
A=matrix(Pol,[[1,1,1],
              [x,1,5],
              [y,2,0]])
```

La ecuación de la recta PO_1 es

$$\begin{vmatrix} 1 & 1 & 1 \\ x & 1 & 5 \\ y & 2 & 0 \end{vmatrix} = 0$$

Por tanto

$$2.00000000000000x + 4.00000000000000y - 10.0000000000000 = 0$$

o equivalentemente

$$x + 2y - 5 = 0$$

b) Distancia de A a la recta PO_1

La distancia del punto $A = (6, 0)$ sobre la recta $x + 2y - 5 = 0$ por tanto es

$$\frac{|6 \cdot 1 + 2 \cdot 0 - 5|}{\sqrt{1^2 + 2^2}} = \frac{1}{\sqrt{5}}$$

c) Coordenadas del punto B.

Las coordenadas del punto B son

```
B=g*vector(RR,[1,6,0])
B=B[1:]
```

$$(5.30901699437495, 0.951056516295154)$$

Obtendríamos el mismo resultado tomando $B = V[1]$.

d) Las coordenadas del punto O_1' centro de la estrella girada.

```
g1O1=g1*vector(RR,[1,5,0])
g1O1=g1O1[1:]
```

Las coordenadas del punto O_1' son

$$O_1' = (5.36722462792756, 2.96298964128541)$$

e) Las coordenadas del punto B', vértice correspondiente con el vértice B de la estrella girada.

```
Bp=g1*g*vector(RR,[1,6,0])
Bp=Bp[1:]
```

Las coordenadas del punto B' son

$$B' = (5.00885667838226, 3.89657006778261)$$

Obtendríamos el mismo resultado tomando $B' = V_1[1]$.

Ejercicio 7. *Determina una estrella regular de 5 puntas con centro en el punto $O_1 = (2, -3)$ y con vértice $A = (3, -2)$. El resto de los vértices se nombran B, C, D y E tal y como indica la figura. Posteriormente se hace una simetría ortogonal respecto de la recta que pasa por los puntos $P = (-2, 7)$ y $Q = (0, 9)$.*



Dibujar dicha estrella así como su simétrica, mostrando además la recta que une los puntos P y Q. Además calcular los siguientes puntos y valores:

a) *La ecuación de la recta PQ.*
b) *La distancia de A a la recta PQ*
c) *Las coordenadas del punto B.*
d) *Las coordenadas del punto O'_1 centro de la estrella simétrica a la dada.*
e) *Las coordenadas del punto B', vértice correspondiente con el vértice B de la estrella simétrica.*

Solución:

El sistema de referencia para realizar el giro y obtener las estrella de vértices A, B, C, D y E, es el que tiene como punto O_1 y base canónica e_1, e_2.

```
R = matrix(RR,[[ 1,0,0],
               [ 2,1,0],
               [-3,0,1]])
R.subdivide(1,1)
```

$$R = \left(\begin{array}{r|rr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline 2.00000000000000 & 1.00000000000000 & 0.000000000000000 \\ -3.00000000000000 & 0.000000000000000 & 1.00000000000000 \end{array}\right)$$

Para obtener los vértices de la estrella, definimos el ángulo $\beta = 2\pi/5$ y la matriz del giro de ángulo β en este sistema de referencia

```
beta = 2*pi/5
G = matrix(RR,[[1, 0         ,  0         ],
               [0, cos(beta), -sin(beta)],
               [0, sin(beta),  cos(beta)]])
g=R*G*R^-1
g.subdivide(1,1)
```

La matriz del giro en sistema de referencia que resulta es

$$g = RG_{\frac{2\pi}{5}}R^{-1} = \left(\begin{array}{r|rr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline -1.47120353763536 & 0.309016994374947 & -0.951056516295154 \\ -3.97506204946546 & 0.951056516295154 & 0.309016994374947 \end{array}\right).$$

Definimos los vértices en homogéneas del pentágono girando el vértice $A = \begin{pmatrix} 3 \\ -2 \end{pmatrix}$

$$V = \{(1,3,-2),\ g(1,3,-2),\ g^2(1,3,-2),\ g^3(1,3,-2),\ g^4(1,3,-2)\}$$

```
V = [g^i*vector(RR,[1,3,-2]) for i in range(5)]
```

$$V[0] = (1.00000000000000,\ 3.00000000000000,\ -2.00000000000000)$$
$$V[1] = (1.00000000000000,\ 1.35796047807979,\ -1.73992648932990)$$
$$V[2] = (1.00000000000000,\ 0.603197753332580,\ -3.22123174208247)$$
$$V[3] = (1.00000000000000,\ 1.77876825791753,\ -4.39680224666742)$$
$$V[4] = (1.00000000000000,\ 3.26007351067010,\ -3.64203952192021)$$

Ahora debemos determinar la simetría respecto de la recta que pasa por los puntos P y Q para obtener los vértices simétricos de los ya calculados.

Para ello tomamos como sistema de referencia un punto cualquiera de la recta junto con una base adecuada. Tomamos como punto, por ejemplo,

$$P = (-2, 7).$$

El primer vector de la base lo tomamos sobre la recta, por ejemplo,

$$v_1 = \vec{PQ} = (0, 9) - (-2, 7) = (2, 2)$$

y el segundo vector un ortogonal, por ejemplo

$$v_2 = (-2, 2).$$

```
P=vector(RR,[-2,7])
Q=vector(RR,[0,9])
v1 = Q-P
v2=vector(RR,[-v1[1],v1[0]])
R=block_matrix(RR,[[       matrix([1,0,0])],
                [column_matrix([P,v1,v2])]])
R.subdivide(1,1)
```

El sistema de referencia es

$$R = \left(\begin{array}{r|rr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline -2.00000000000000 & 2.00000000000000 & -2.00000000000000 \\ 7.00000000000000 & 2.00000000000000 & 2.00000000000000 \end{array}\right)$$

Introducimos la matriz de simetría:

```
sRR = matrix(RR,[[1,0, 0],
              [0,1, 0],
              [0,0,-1]])
sRR.subdivide(1,1)
s=R*sRR*R^-1
s.subdivide(1,1)
```

Respecto de dicho sistema de referencia la matriz de la simetría es

$$S_{RR} = \left(\begin{array}{r|rr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline 0.000000000000000 & 1.00000000000000 & 0.000000000000000 \\ 0.000000000000000 & 0.000000000000000 & -1.00000000000000 \end{array}\right)$$

Y respecto del sistema canónico es la matriz

$$s = RS_{RR}R^{-1} = \left(\begin{array}{r|rr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline -9.00000000000000 & 0.000000000000000 & 1.00000000000000 \\ 9.00000000000000 & 1.00000000000000 & 0.000000000000000 \end{array}\right)$$

Los vértices de la estrella simétrica son

$$V = \{s(1,3,-2),\ s(g(1,3,-2)),\ s(g^2(1,3,-2)),\ s(g^3(1,3,-2)),\ s(g^4(1,3,-2))\}$$

```
V1 = [s*v for v in V]
```

$$V_1[0] = (1.00000000000000,\ -11.0000000000000,\ 12.0000000000000)$$
$$V_1[1] = (1.00000000000000,\ -10.7399264893299,\ 10.3579604780798)$$
$$V_1[2] = (1.00000000000000,\ -12.2212317420825,\ 9.60319775333258)$$
$$V_1[3] = (1.00000000000000,\ -13.3968022466674,\ 10.7787682579175)$$
$$V_1[4] = (1.00000000000000,\ -12.6420395219202,\ 12.2600735106701)$$

Quitamos la coordenada homogénea para poder representar los puntos en el plano

```
V = [vector(v[1:]) for v in V]
V1 = [vector(v[1:]) for v in V1]
```

$$V[0] = (3.00000000000000,\ -2.00000000000000) \quad V_1[0] = (-11.0000000000000,\ 12.0000000000000)$$
$$V[1] = (1.35796047807979,\ -1.73992648932990) \quad V_1[1] = (-10.7399264893299,\ 10.3579604780798)$$
$$V[2] = (0.603197753332580,\ -3.22123174208247) \quad V_1[2] = (-12.2212317420825,\ 9.60319775333258)$$
$$V[3] = (1.77876825791753,\ -4.39680224666742) \quad V_1[3] = (-13.3968022466674,\ 10.7787682579175)$$
$$V[4] = (3.26007351067010,\ -3.64203952192021) \quad V_1[4] = (-12.6420395219202,\ 12.2600735106701)$$

Dibujamos las estrellas así como los puntos P y Q, el segmento que los une y los centros O_1 y O_1'.

```
O1=vector(RR,[2,-3])
sO1=s*vector(RR,[1,2,-3])
sO1=sO1[1:]
```

De esta forma,

$$O_1 = (2.00000000000000,\ -3.00000000000000)$$
$$O_1' = s(O_1) = (-12.0000000000000,\ 11.0000000000000)$$

Obteniendo:



El código que hay tras el dibujo es el siguiente:

```
\begin{sagesub}
\begin{center}
\begin{tikzpicture}[scale = 0.5,
cara1/.style={thick, color = blue, fill opacity = 0.3, fill = blue!20},
cara2/.style={thick, color = red, fill opacity = 0.3, fill = red!20}]
\draw[->,thick,gray] (4.5,0) -- (-14,0); % Eje X
\draw[->,thick,gray] (0,-5) -- (0,14); % Eje Y
```

```
\draw[cara1] !{V[0]} -- !{V[2]} -- !{V[4]} -- !{V[1]} -- !{V[3]} -- cycle;
\draw[cara2] !{V1[0]} -- !{V1[2]} -- !{V1[4]} -- !{V1[1]} -- !{V1[3]} -- cycle;
\draw[dashed] (-9,0) -- (2,11);
\draw[dashed] !{O1} -- !{sO1};
\draw[fill = white] !{P} circle (0.5mm);
\draw[fill = white] !{Q} circle (0.5mm);
\draw[fill = white] !{P} circle (0.5mm) node[above left] {$P$};
\draw[fill = white] !{Q} circle (0.5mm) node[above left] {$Q$};
\draw[fill = white] !{O1} circle (0.5mm);
\draw[fill = white] !{sO1} circle (0.5mm);
\end{tikzpicture}
\end{center}
\end{sagesub}
```

Además

a) Ecuación de la recta PQ.

```
Pol.<x,y>=PolynomialRing(RR)
A=matrix(Pol,[[1,1,1],
              [x,-2,0],
              [y,7,9]])
```

La ecuación de la recta PQ es

$$\begin{vmatrix} 1 & 1 & 1 \\ x & -2 & 0 \\ y & 7 & 9 \end{vmatrix} = 0$$

Por tanto

$$-2.00000000000000x + 2.00000000000000y - 18.0000000000000 = 0$$

o equivalentemente

$$-x + y - 9 = 0$$

b) Distancia de A a la recta PQ

La distancia del punto $A = (3, -2)$ sobre la recta $-x + y - 9 = 0$ por tanto es

$$\frac{|-(3) + (-2) - 9|}{\sqrt{(-1)^2 + 1^2}} = \frac{14}{\sqrt{2}} = 7\sqrt{2}$$

c) Coordenadas del punto B.

Las coordenadas del punto B son

```
B=g*vector(RR,[1,3,-2])
B=B[1:]
```

$$B = (1.35796047807979, -1.73992648932990)$$

Obtendríamos el mismo resultado tomando $B = V[1]$.

d) Las coordenadas del punto O_1' centro de la estrella simétrica a la dada.

```
sO1=s*vector(RR,[1,2,-3])
sO1=sO1[1:]
```

Las coordenadas del punto O_1' son

$$O_1' = (-12.0000000000000, 11.0000000000000)$$

e) Las coordenadas del punto B', vértice correspondiente con el vértice B de la estrella simétrica.

```
Bp=s*g*vector(RR,[1,3,-2])
Bp=Bp[1:]
```

Las coordenadas del punto B' son

$$B' = (-10.7399264893299, 10.3579604780798)$$

Obtendríamos el mismo resultado tomando $B' = V_1[1]$.

Ejercicio 8. *Calcula la matriz de la transformación afín que corresponde a la simetría ortogonal con respecto al plano que pasa por los puntos* $A = \begin{bmatrix} 1 \\ 2 \\ -1 \end{bmatrix}$, $B = \begin{bmatrix} 2 \\ 1 \\ 1 \end{bmatrix}$ *y* $C = \begin{bmatrix} -1 \\ 3 \\ 1 \end{bmatrix}$.

Solución:

```
A = vector(RR, [1, 2, -1])
B = vector(RR, [2, 1, 1])
C = vector(RR, [-1, 3, 1])

AB = B - A
AC = C - A
n = AB.cross_product(AC)

u1 = AB.normalized()
u2 = n.cross_product(AB)
u3 = n.normalized()

R = block_matrix(2, 1, [matrix(RR, [1, 0, 0, 0]), column_matrix([A, u1, u2, u3])])
R.subdivide(1, 1)

MRRs = matrix(RR, [[1, 0, 0, 0], [0, 1, 0, 0], [0, 0, 1, 0], [0, 0, 0, -1]])

Ms = R * MRRs * R^-1
Ms.subdivide(1, 1)
```

$$M_s = \left(\begin{array}{r|rrr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline 2.26415094339623 & 0.396226415094339 & -0.905660377358490 & -0.150943396226415 \\ 3.39622641509434 & -0.905660377358490 & -0.358490566037736 & -0.226415094339623 \\ 0.566037735849057 & -0.150943396226415 & -0.226415094339623 & 0.962264150943396 \end{array}\right)$$

Ejercicio 9. *Calcula la matriz de la homotecia con factor* $\frac{1}{2}$ *y con centro en el punto* $\begin{bmatrix} -2 \\ 2 \\ 1 \end{bmatrix}$

Solución:

```
R = matrix(RR, [[1, 0, 0, 0], [-2, 1, 0, 0], [2, 0, 1, 0], [1, 0, 0, 1]])
f = 1/2
MRRh = matrix(RR, [[1, 0, 0, 0], [0, f, 0, 0], [0, 0, f, 0], [0, 0, 0, f]])
Mh = R * MRRh * R^-1
Mh.subdivide(1, 1)
```

$$M_h = \left(\begin{array}{r|rrr} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline -1.00000000000000 & 0.500000000000000 & 0.000000000000000 & 0.000000000000000 \\ 1.00000000000000 & 0.000000000000000 & 0.500000000000000 & 0.000000000000000 \\ 0.500000000000000 & 0.000000000000000 & 0.000000000000000 & 0.500000000000000 \end{array}\right)$$

Ejercicio 10. *Calcula los vértices de un heptágono regular en $\mathcal{A}^3(\mathbb{R})$, sabiendo que están contenidos en el plano $3x + y - z = 1$, que su centro está en el punto $P = \begin{bmatrix} 1 \\ 0 \\ 2 \end{bmatrix}$, y que uno de dichos vértices es el punto $Q = \begin{bmatrix} 1 \\ 1 \\ 3 \end{bmatrix}$.*

Solución.-

```
n = vector(RR, [3, 1, -1])
P = vector(RR, [1, 0, 2])
Q = vector(RR, [1, 1, 3])

PQ = Q - P

u1 = PQ.normalized()
u2 = PQ.cross_product(n).normalized()
u3 = n.normalized()

R = block_matrix(2, 1, [matrix(RR, [1, 0, 0, 0]), column_matrix([P, u1, u2, u3])])
R.subdivide(1, 1)

a = 2*pi/7
MRRg = matrix(RR, [[1, 0, 0, 0], [0, cos(a), -sin(a), 0], [0, sin(a), cos(a), 0], [0, 0, 0, 1]])

Mg = R * MRRg * R^-1

C = [Mg^i * vector(RR, [1, 1, 1, 3]) for i in range(7)]
W = [v[1:] for v in C]
```



Ejercicio 11. *Calcula el simétrico de cada uno de los puntos*

$$\begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix} \quad \begin{bmatrix} 2 \\ -1 \\ -3 \end{bmatrix} \quad \begin{bmatrix} 3 \\ 3 \\ -1 \end{bmatrix} \quad \begin{bmatrix} 2 \\ 3 \\ 1 \end{bmatrix} \quad \begin{bmatrix} 4 \\ -6 \\ 10 \end{bmatrix} \quad \begin{bmatrix} 5 \\ 10 \\ -15 \end{bmatrix}$$

respecto del plano $x + y = 1$. Dibuja todos los puntos, sus simétricos y el plano.

Solución:

```
D = vector(RR, [1, 1, 1, 0]), vector(RR, [1, 2, -1, -3]), vector(RR, [1, 3, -3, -1]), vector(RR, [1, 2, 3,

n = vector(RR, [1, 1, 0])
v1 = vector(RR, [-n[1], n[0], 0])
v2 = n.cross_product(v1)

u1, u2, u3 = v1.normalized(), v2.normalized(), n.normalized()

# x = 0, y = 1
R = block_matrix(2, 1, [matrix(RR, [1, 0, 0, 0]), column_matrix(RR, [vector(RR, [0, 1, 0]), u1, u2, u3])])

MRRs = matrix(RR, [[1, 0, 0, 0], [0, 1, 0, 0],  [0, 0, 1, 0], [0, 0, 0, -1]])

Ms = R * MRRs * R^-1

sime = [Ms * v for v in D]
```

El código que hay tras el dibujo es:

$$\begin{pmatrix} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 & 0.000000000000000 \\ 1.00000000000000 & 0.000000000000000 & -1.00000000000000 & 0.000000000000000 \\ 1.00000000000000 & -1.00000000000000 & 0.000000000000000 & 0.000000000000000 \\ 0.000000000000000 & 0.000000000000000 & 0.000000000000000 & 1.00000000000000 \end{pmatrix}$$



Ejercicio 12. *Sabiendo que la simetría ortogonal $f : \mathcal{A}^3(\mathbb{R}) \to \mathcal{A}^3(\mathbb{R})$ respecto a un cierto plano cumple*

$$f\left(\begin{bmatrix} 1 \\ 3 \\ -1 \end{bmatrix}\right) = \begin{bmatrix} 3 \\ 2 \\ 1 \end{bmatrix},$$

calcula un sistema de referencia afín de dicho plano y la matriz de f respecto del sistema de referencia canónico de $\mathcal{A}^3(\mathbb{R})$.

Solución:

Calcular plano

```
A = vector(RR, [1, 3, -1])
B = vector(RR, [3, 2, 1])
AB = B - A
PM = A + (AB / 2)

n = AB
v1 = vector(RR, [-n[1], n[0], 0])
v2 = n.cross_product(v1)


u1 = v1.normalized()
u2 = v2.normalized()
u3 = n.normalized()

R = block_matrix(2, 1, [matrix(RR, [1, 0, 0, 0]), column_matrix([PM, u1, u2, u3])])

MRRs = matrix(RR, [[1, 0, 0, 0], [0, 1, 0, 0], [0, 0, 1, 0], [0, 0, 0, -1]])

Ms = R * MRRs * R^-1
```

$$\begin{pmatrix} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 & 0.000000000000000 \\ 0.666666666666667 & 0.111111111111111 & 0.444444444444444 & -0.888888888888889 \\ -0.333333333333333 & 0.444444444444444 & 0.777777777777778 & 0.444444444444444 \\ 0.666666666666667 & -0.888888888888889 & 0.444444444444444 & 0.111111111111111 \end{pmatrix}$$



Ejercicio 13. *Dibuja un prisma en $\mathcal{A}^3(\mathbb{R})$ con centro de gravedad el origen de coordenadas y bases pentagonales regulares en planos paralelos al plano $\Pi \equiv x + y + 2z = 0$, y que tenga al punto $A = \begin{bmatrix} 0.5 \\ 0.1 \\ 0.4 \end{bmatrix}$ como uno de sus vértices.*

Solución:

```
# plano
pn = vector(RR, [1, 1, 2])
v1 = vector(RR, [-pn[1], pn[0], 0])
v2 = pn.cross_product(v1)

A = vector(RR, [0.5, 0.1, 0.4])
Ah = vector(RR, [1, 00.5, 0.1, 0.4])

# proyectar A en la normal del plano obteniendo centro de cara
u1, u2, u3 = pn.normalized(), v1.normalized(), v2.normalized()
pB = column_matrix([u1, u2, u3])
MBBpv = matrix(RR, [[1, 0, 0], [0, 0, 0], [0, 0, 0]])
Mpn = pB * MBBpv * pB^-1
cO = Mpn * A

# rotar A en el centro de la cara obteniendo puntos de cara
cR = block_matrix(2, 1, [matrix(RR, [1, 0, 0, 0]), column_matrix([cO, u2, u3, u1])])
a = 2*pi/5
McRcRr = matrix(RR, [[1, 0, 0, 0], [0, cos(a), -sin(a), 0], [0, sin(a), cos(a), 0], [0, 0, 0, 1]])
Mcr = cR * McRcRr * cR^-1
cAP = [Mcr^i * Ah for i in range(5)]
cAP = [v[1:] for v in cAP]

# simetria de cara por el plano obteniendo la otra cara
MBBsp = matrix(RR, [[-1, 0, 0], [0, 1, 0], [0, 0, 1]])
Msp = pB * MBBsp * pB^-1
cBP = [Msp * v for v in cAP]
```


r_2
r_3
O_c
r_1
u_3
A
u_1
u_2

Ejercicio 14. *Calcula los vértices de una* PIRÁMIDE $C \subseteq \mathcal{A}^3(\mathbb{R})$ *de altura* 5 *cuya base sea un pentágono regular inscrito en una circunferencia de radio* 1 *que esté contenido en el plano de ecuación* $x - y + z = 2$.

Solución:

Tomamos como punto base, un punto del plano $P = \begin{pmatrix} 1 \\ 0 \\ 1 \end{pmatrix}$ y como base, una base $B = \{v_1, v_2, v_3\}$ donde $\{v_1, v_2\}$ sea una base ortonormal del espacio de dirección del plano y el vector v_3 un vector unitario ortogonal al plano.

```
P=vector(RR,[1,0,1])
v3 = vector(RR,[1,-1,1])
v1 = vector(RR,[1,1,0])
v2 = v1.cross_product(v3)
v1=v1/norm(v1)
v2=v2/norm(v2)
v3=v3/norm(v3)
```

Sistema de referencia R:

```
R = block_matrix(RR,[[       matrix([1, 0, 0, 0])],
                      [column_matrix([P,v1,v2,v3])]])
R.subdivide(1,1)
```

De donde

$$R = \left(\begin{array}{c|ccc} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline 1.00000000000000 & 0.707106781186547 & 0.408248290463863 & 0.577350269189626 \\ 0.000000000000000 & 0.707106781186547 & -0.408248290463863 & -0.577350269189626 \\ 1.00000000000000 & 0.000000000000000 & -0.816496580927726 & 0.577350269189626 \end{array}\right).$$

Para calcular los puntos de la base de la pirámide construimos la matriz del giro de ángulo $2\pi/5$ y la pasamos al nuevo sistema de referencia.

```
a = 2*pi/5
G = column_matrix(RR,[[1,    0,      0,  0],
                      [0,cos(a),sin(a), 0],
                      [0,-sin(a),cos(a),0],
                      [0,    0,      0,  1]])
g = R*G*R^-1
g.subdivide(1,1)
```

$$g = RG_{\frac{2\pi}{5}}R^{-1} = \left(\begin{array}{c|ccc} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline -0.318765067155870 & 0.539344662916632 & 0.318765067155871 & 0.779420404239239 \\ 0.460655337083368 & -0.779420404239239 & 0.539344662916632 & 0.318765067155870 \\ 0.779420404239239 & -0.318765067155871 & -0.779420404239239 & 0.539344662916632 \end{array}\right)$$

Los vértices de la pirámide, pentágono regular, está inscrito en una circunferencia de radio 1. Nuestra base es ortonormal. Por tanto, $||v_1|| = 1$. Así que podemos suponer por tanto que el punto que debemos girar es aquel que tiene como coordenadas $[1, 1, 0, 0]$ en el sistema de referencia R. Dicho punto en el sistema de referencia canónico es

$$R * vector(RR, [1, 1, 0, 0])$$

o equivalentemente el punto $P + v_1$.

Los puntos de la base de la pirámide serán los que corresponden a girar dicho punto cinco veces.

```
puntogiro=R*vector(RR,[1,1,0,0])
BasePiramideh = [g^i*puntogiro for i in range(5)]
```

El vértice de la pirámide, es el punto que tiene como coordenadas $[1, 0, 0, 5]$ respecto del sistema de referencia R (punto de altura 5). Dicho punto en el sistema de referencia canónico es

$$R * vector(RR, [1, 0, 0, 5])$$

```
Verticeh = R*vector(RR,[1,0,0,5])
```

Para poder dibujar la pirámide eliminamos la coordenada homogénea y unimos las aristas de la base y las que unen el vértice con los vértices de la base. Finalmente generamos el dibujo.

```
BP = [v[1:] for v in BasePiramideh]
Vertice = Verticeh[1:]
```

Los vértices base de la pirámide son:

$$BP[0] = (1.70710678118655, 0.707106781186547, 1.00000000000000)$$
$$BP[1] = (1.60677520913642, -0.169759184687603, 0.223465606175973)$$
$$BP[2] = (0.667900921590589, -0.812023727225957, 0.520075351183455)$$
$$BP[3] = (0.187976272774043, -0.332099078409412, 1.47992464881655)$$
$$BP[4] = (0.830240815312397, 0.606775209136423, 1.77653439382403)$$

Y el vértice de la pirámide es:

$$Vertice = (3.88675134594813, -2.88675134594813, 3.88675134594813)$$

Si ejecutamos el siguiente código de Sagesub:

```
\begin{sagesub}
\begin{center}
\tdplotsetmaincoords{70}{110}
\begin{tikzpicture}[scale = 1.6, tdplot_main_coords,
  cara/.style={thick, color = blue, fill opacity = 0.3, fill = blue!20},
  ]
\draw[->,thick,gray] (-2.5,0,0) -- (3.5,0,0); % Eje X
\draw[->,thick,gray] (0,-2.5,0) -- (0,2.5,0); % Eje Y
\draw[->,thick,gray] (0,0,-2.5) -- (0,0,2.5); % Eje Z
\draw[cara] !{BP[0]} -- !{BP[1]} -- !{BP[2]} -- !{BP[3]} -- !{BP[4]} --cycle;
\draw[cara] !{BP[0]} -- !{Vertice} -- !{BP[1]} -- cycle;
\draw[cara] !{BP[1]} -- !{Vertice} -- !{BP[2]} -- cycle;
\draw[cara] !{BP[2]} -- !{Vertice} -- !{BP[3]} -- cycle;
\draw[cara] !{BP[3]} -- !{Vertice} -- !{BP[4]} -- cycle;
\draw[cara] !{BP[4]} -- !{Vertice} -- !{BP[0]} -- cycle;
\end{tikzpicture}
\end{center}
\end{sagesub}
```

obtenemos:



Ejercicio 15. *Calcula los vértices de una* TORRE $C \subseteq \mathcal{A}^3(\mathbb{R})$ *de altura* 3 *cuya base sea un hexágono regular de lado* 1 *que esté contenido en el plano de ecuación* $2x - y + z = 1$.

Solución.-

Para construir los vectores de la base, tomaremos v_3 el vector normal al plano y $\{v_1, v_2\}$ una base ortonormal del espacio de direcciones del plano.

```
v3 = vector(RR,[2,-1,1])
v1 = vector(RR,[1,2,0])
v2 = v1.cross_product(v3)
v1=v1/norm(v1)
v2=v2/norm(v2)
v3=v3/norm(v3)
```

El sistema de referencia que debemos tomar tiene como punto base, un punto del plano $P = \begin{pmatrix} 1 \\ 1 \\ 0 \end{pmatrix}$ y como vectores los que hemos calculado, v_1, v_2 y v_3.

```
P = vector(RR,[1,1,0])
R = block_matrix(RR,[[       matrix([1, 0, 0, 0])],
                     [column_matrix([P,v1,v2,v3])]])
R.subdivide(1,1)
```

$$R = \left(\begin{array}{c|ccc} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline 1.00000000000000 & 0.447213595499958 & 0.365148371670111 & 0.816496580927726 \\ 1.00000000000000 & 0.894427190999916 & -0.182574185835055 & -0.408248290463863 \\ 0.000000000000000 & 0.000000000000000 & -0.912870929175277 & 0.408248290463863 \end{array}\right)$$

Para calcular los puntos de la base de la torre construimos la matriz del giro de ángulo $2\pi/6$ y la pasamos al nuevo sistema de referencia.

```
a = 2*pi/6
G = column_matrix(RR,[[1,    0,      0,  0],
                     [0,cos(a),sin(a), 0],
                     [0,-sin(a),cos(a),0],
                     [0,    0,      0,  1]])
g = R*G*R^-1
g.subdivide(1,1)
```

$$g = RG_{\frac{2\pi}{6}}R^{-1} = \left(\begin{array}{c|ccc} 1.00000000000000 & 0.000000000000000 & 0.000000000000000 & 0.000000000000000 \\ \hline -0.0202200572599402 & 0.833333333333333 & 0.186886723926607 & 0.520220057259940 \\ 0.936886723926607 & -0.520220057259941 & 0.583333333333333 & 0.623773447853214 \\ 0.977326838446488 & -0.186886723926607 & -0.790440114519881 & 0.583333333333333 \end{array}\right)$$

Los vértices de la base de la torre, hexágono regular, está inscrito en una circunferencia de radio 1. Los hexágonos regulares tienen la propiedad de que la distancia del centro del hexágono a cualquier vértice del mismo mide lo mismo que la arista. Por lo tanto, podemos suponer por tanto que el punto que debemos girar es aquel que tiene como coordenadas $[1, 1, 0, 0]$ en el sistema de referencia R. Dicho punto en el sistema de referencia canónico es

$$R * vector(RR, [1, 1, 0, 0])$$

o equivalentemente el punto $P + v_1$.

```
BaseTorreh = [g^i*R*vector(RR,[1,1,0,0]) for i in range(6)]
```

Los puntos de la base de la torre serán los que corresponden a girar el primero de los puntos seis veces.

Para obtener la tapa de la torre, sumamos a cada punto de la base de la torre el vector de coordenadas $[1, 0, 0, 3]$ en el sistema de referencia R (vector de altura 3). Es decir le sumamos el vector

$$R * vector[1, 0, 0, 3]$$

```
TapaTorreh = [v+R*vector(RR,[1,0,0,3]) for v in BaseTorreh]
```

Seguidamente eliminamos la coordenada homogénea y unimos las aristas de la base, las aristas de la tapa además de las aristas laterales de la torre. Finalmente generamos el dibujo.

```
BT = [v[1:] for v in BaseTorreh]
TT = [w[1:] for w in TapaTorreh]
```

Podemos entonces introducir el siguiente código en sagesub:

```
\begin{sagesub}
\begin{center}
\tdplotsetmaincoords{70}{110}
\begin{tikzpicture}[scale = 2.4, tdplot_main_coords,
  cara/.style={thick, color = blue, fill opacity = 0.3, fill = blue!20},
  ]
\draw[->,thick,gray] (-2.5,0,0) -- (3.5,0,0); % Eje X
\draw[->,thick,gray] (0,-2.5,0) -- (0,2.5,0); % Eje Y
\draw[->,thick,gray] (0,0,-2.5) -- (0,0,2.5); % Eje Z
\draw[cara] !{BT[0]} -- !{BT[1]} -- !{BT[2]} -- !{BT[3]} -- !{BT[4]} -- !{BT[5]} --cycle;
\draw[cara] !{TT[0]} -- !{TT[1]} -- !{TT[2]} -- !{TT[3]} -- !{TT[4]} -- !{TT[5]} --cycle;
\draw[cara] !{BT[0]} -- !{TT[0]} -- !{TT[1]} -- !{BT[1]} -- cycle;
\draw[cara] !{BT[1]} -- !{TT[1]} -- !{TT[2]} -- !{BT[2]} -- cycle;
\draw[cara] !{BT[2]} -- !{TT[2]} -- !{TT[3]} -- !{BT[3]} -- cycle;
\draw[cara] !{BT[3]} -- !{TT[3]} -- !{TT[4]} -- !{BT[4]} -- cycle;
\draw[cara] !{BT[4]} -- !{TT[4]} -- !{TT[5]} -- !{BT[5]} -- cycle;
\draw[cara] !{BT[5]} -- !{TT[5]} -- !{TT[0]} -- !{BT[0]} -- cycle;
\end{tikzpicture}
\end{center}
\end{sagesub}
```

para realizar el dibujo que nos piden: