

## AMD CURSO 2023-2024. PRÁCTICAS DE LA SEMANA 8. ECUACIONES MATRICIALES

### Cuestiones teóricas previas

Cuando realizamos operaciones elementales por filas sobre la matriz ampliada  $[A|b]$  de un sistema de ecuaciones  $Ax = b$ , pasando de  $[A|b]$  a  $R = [N|m]$ , transformamos el sistema original en otro equivalente,  $Nx = m$ , que es siempre más sencillo de resolver.

- Si  $N = I$ , el sistema es compatible determinado y  $x = m$ .
- Si  $N \neq I$ , el carácter del sistema (será bien compatible indeterminado o bien incompatible) es el que tenga el nuevo sistema  $Nx = m$ .

**Ejercicio 1.** *Resuelve el siguiente sistema de ecuaciones sobre  $\mathbb{Z}_5$*

```
A = matrix(Zmod(5), [[4,2,1,0,1,4],
[1,3,4,0,0,1],
[4,2,2,2,0,4],
[3,4,3,2,2,3],
[1,3,1,4,0,1]])
b = matrix(Zmod(5), [[4],[0],[4],[2],[3]])
Pol = PolynomialRing(Zmod(5), x, 6)
x = column_matrix(Pol, Pol.gens())
Ecuaciones = A*x
```

$$\begin{aligned} -x_0 + 2x_1 + x_2 + x_4 - x_5 &= 4 \\ x_0 + 3x_1 - x_2 + x_5 &= 0 \\ -x_0 + 2x_1 + 2x_2 + 2x_3 - x_5 &= 4 \\ 3x_0 - x_1 + 3x_2 + 2x_3 + 2x_4 + 3x_5 &= 2 \\ x_0 + 3x_1 + x_2 - x_3 + x_5 &= 3 \end{aligned}$$

*Solución:*

```
Ab = block_matrix([[A, b]])
Abr = Ab.echelon_form()
Abr = copy(Abr)
Abr.subdivide(3, 6)
```

$$[A|b] = \left( \begin{array}{cccccc|c} 4 & 2 & 1 & 0 & 1 & 4 & 4 \\ 1 & 3 & 4 & 0 & 0 & 1 & 0 \\ 4 & 2 & 2 & 2 & 0 & 4 & 4 \\ 3 & 4 & 3 & 2 & 2 & 3 & 2 \\ 1 & 3 & 1 & 4 & 0 & 1 & 3 \end{array} \right) \rightarrow \left( \begin{array}{cccccc|c} 1 & 3 & 0 & 2 & 0 & 1 & 4 \\ 0 & 0 & 1 & 2 & 0 & 0 & 4 \\ 0 & 0 & 0 & 0 & 1 & 0 & 4 \\ \hline 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{array} \right)$$

$$x_0 + 3x_1 + 2x_3 + x_5 = 4$$

$$x_2 + 2x_3 = 4$$

$$x_4 = 4$$

Los elementos no pivote son parametros libres

$$x_0 = 4 - 3\alpha - 2\beta - \gamma$$

$$x_1 = \alpha$$

$$x_2 = 4 - 2\beta$$

$$x_3 = \beta$$

$$x_4 = 4$$

$$x_5 = \gamma$$

### Cuestiones teóricas previas

Supongamos que queremos resolver una ecuación matricial del tipo

$$A \cdot X = B$$

Para ello, podemos ampliar la matriz  $A$  por la izquierda con la matriz identidad  $I$  de tamaño adecuado y aplicar a la matriz ampliada el método de reducción de Gauss:

$$[A|I] \rightarrow R = [N|P]$$

A la matriz  $P$  la llamamos “matriz de paso”. Ella cumple la igualdad:  $P \cdot A = N$  y, como las operaciones elementales por filas se pueden deshacer, existe una matriz  $P^*$  tal que  $P^* \cdot P \cdot A = A$ . Por tanto, la ecuación original,  $A \cdot X = B$ , es equivalente a la nueva ecuación

$$P \cdot A \cdot X = P \cdot B,$$

que podemos reescribir como:

$$N \cdot X = P \cdot B$$

Esta nueva ecuación matricial siempre es más sencilla de resolver que la ecuación original.

**Ejercicio 2.** Calcular todas las matrices  $X$  sobre el cuerpo  $\mathbb{Z}_5$  tales que

$$\begin{pmatrix} 4 & 2 & 4 & 1 \\ 2 & 4 & 4 & 3 \\ 1 & 3 & 0 & 1 \end{pmatrix} X = \begin{pmatrix} 0 & 4 \\ 1 & 1 \\ 1 & 0 \end{pmatrix}.$$

*Solución:*

Introducimos las matrices  $A$  y  $B$  como sigue

```
A = matrix(Zmod(5), [[4,2,4,1], [2,4,4,3], [1,3,0,1]])
```

```
B = matrix(Zmod(5), [[0,4], [1,1], [1,0]])
```

donde  $A = \begin{pmatrix} 4 & 2 & 4 & 1 \\ 2 & 4 & 4 & 3 \\ 1 & 3 & 0 & 1 \end{pmatrix}$  y  $B = \begin{pmatrix} 0 & 4 \\ 1 & 1 \\ 1 & 0 \end{pmatrix}$ .

Determinar el tamaño de la matriz  $X$  es sencillo, ya que  $X$  deberá tener 4 filas para poder multiplicarse por la izquierda con la matriz que nos proponen y dos columnas para que el resultado sea una matriz  $3 \times 2$ . Es decir,  $X$  es una matriz de incógnitas de 4 filas y 2 columnas. La definimos en SageMath:

```
Pol = PolynomialRing(Zmod(5), 8, 'x')
```

```
X = matrix(Pol, 4, 2, Pol.gens())
```

El sistema que tenemos es por lo tanto:

$$\begin{pmatrix} 4 & 2 & 4 & 1 \\ 2 & 4 & 4 & 3 \\ 1 & 3 & 0 & 1 \end{pmatrix} \begin{pmatrix} x_0 & x_1 \\ x_2 & x_3 \\ x_4 & x_5 \\ x_6 & x_7 \end{pmatrix} = \begin{pmatrix} 0 & 4 \\ 1 & 1 \\ 1 & 0 \end{pmatrix}$$

Para resolverlo vamos a reducir la matriz  $A$  obteniendo la matriz de paso  $P$ .

```
M = block_matrix([[A,1]])
```

```
R = M.echelon_form()
```

$$[A|I] = \left( \begin{array}{cccc|ccc} 4 & 2 & 4 & 1 & 1 & 0 & 0 \\ 2 & 4 & 4 & 3 & 0 & 1 & 0 \\ 1 & 3 & 0 & 1 & 0 & 0 & 1 \end{array} \right) \rightarrow R = [N|P] = \left( \begin{array}{cccc|ccc} 1 & 0 & 0 & 2 & 1 & 4 & 4 \\ 0 & 1 & 0 & 3 & 3 & 2 & 4 \\ 0 & 0 & 1 & 3 & 4 & 0 & 4 \end{array} \right)$$

Obtenemos entonces la matriz de paso:

```
P = R.subdivision(0,1)
PA = P*A
PB = P*B
```

$$P = \begin{pmatrix} 1 & 4 & 4 \\ 3 & 2 & 4 \\ 4 & 0 & 4 \end{pmatrix}$$

y las matrices:

$$PA = \begin{pmatrix} 1 & 4 & 4 \\ 3 & 2 & 4 \\ 4 & 0 & 4 \end{pmatrix} \begin{pmatrix} 4 & 2 & 4 & 1 \\ 2 & 4 & 4 & 3 \\ 1 & 3 & 0 & 1 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & 3 \\ 0 & 0 & 1 & 3 \end{pmatrix}$$

y

$$PB = \begin{pmatrix} 1 & 4 & 4 \\ 3 & 2 & 4 \\ 4 & 0 & 4 \end{pmatrix} \begin{pmatrix} 0 & 4 \\ 1 & 1 \\ 1 & 0 \end{pmatrix} = \begin{pmatrix} 3 & 3 \\ 1 & 4 \\ 4 & 1 \end{pmatrix}$$

De esta forma el sistema  $AX = B$  es equivalente a:

$$PA \cdot X = PB \Leftrightarrow \begin{pmatrix} 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & 3 \\ 0 & 0 & 1 & 3 \end{pmatrix} \begin{pmatrix} x_0 & x_1 \\ x_2 & x_3 \\ x_4 & x_5 \\ x_6 & x_7 \end{pmatrix} = \begin{pmatrix} 3 & 3 \\ 1 & 4 \\ 4 & 1 \end{pmatrix}$$

A la matriz reducida le falta un pivote y si lo añadimos, tenemos que añadir una nueva fila a la parte derecha, que debemos rellenar con parámetros libres:

```
Pivotes = matrix(Zmod(5), [[0,0,0,1]])
Pol2.<a,b> = PolynomialRing(Zmod(5))
Parametros = matrix(Pol2, [[a,b]])
PAE = block_matrix([[PA],[Pivotes]])
PBE = block_matrix([[PB],[Parametros]])
```

$$\begin{pmatrix} 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & 3 \\ 0 & 0 & 1 & 3 \\ 0 & 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} x_0 & x_1 \\ x_2 & x_3 \\ x_4 & x_5 \\ x_6 & x_7 \end{pmatrix} = \begin{pmatrix} 3 & 3 \\ 1 & 4 \\ 4 & 1 \\ a & b \end{pmatrix}$$

Ahora la matriz que multiplica a  $X$  ya es invertible, porque es cuadrada, está en forma escalonada reducida, y tiene todos sus pivotes. Por tanto, podemos despejar:

$$\begin{aligned} \begin{pmatrix} x_0 & x_1 \\ x_2 & x_3 \\ x_4 & x_5 \\ x_6 & x_7 \end{pmatrix} &= \begin{pmatrix} 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & 3 \\ 0 & 0 & 1 & 3 \\ 0 & 0 & 0 & 1 \end{pmatrix}^{-1} \begin{pmatrix} 3 & 3 \\ 1 & 4 \\ 4 & 1 \\ a & b \end{pmatrix} = \\ &= \begin{pmatrix} 1 & 0 & 0 & 3 \\ 0 & 1 & 0 & 2 \\ 0 & 0 & 1 & 2 \\ 0 & 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} 3 & 3 \\ 1 & 4 \\ 4 & 1 \\ a & b \end{pmatrix} = \begin{pmatrix} 3a+3 & 3b+3 \\ 2a+1 & 2b+4 \\ 2a+4 & 2b+1 \\ a & b \end{pmatrix} \end{aligned}$$

donde  $a, b \in \mathbb{Z}_5$  son parámetros libres. Esto resuelve el problema.

### Cuestiones teóricas previas

Supongamos que queremos resolver una ecuación matricial del tipo:

$$X \cdot C = B$$

Tomando traspuestas, dicha ecuación se transforma en la ecuación matricial equivalente:

$$C^T \cdot X^T = B^T$$

Por tanto, si multiplicamos por la matriz de paso,  $P$ , de  $C^T$ , el sistema es también equivalente a

$$P \cdot C^T \cdot X^T = P \cdot B^T$$

Esta nueva ecuación matricial tendrá la característica de ser más sencilla de resolver que la original.

**Ejercicio 3.** *Calcula todas las matrices  $X$  sobre  $\mathbb{Z}_5$  tales que  $XC = B$  siendo*

```
B = matrix(Zmod(5), [[1,4,1],
[3,3,0]])
C = matrix(Zmod(5), [[0,1,2],
[1,2,2],
[1,1,0]])
```

$$C = \begin{pmatrix} 0 & 1 & 2 \\ 1 & 2 & 2 \\ 1 & 1 & 0 \end{pmatrix} \quad B = \begin{pmatrix} 1 & 4 & 1 \\ 3 & 3 & 0 \end{pmatrix}$$

*Solución:*

Introducimos las matrices  $C^T$  y  $B^T$  como sigue

```
CT = C.T
BT = B.T
```

donde  $C^T = \begin{pmatrix} 0 & 1 & 1 \\ 1 & 2 & 1 \\ 2 & 2 & 0 \end{pmatrix}$  y  $B^T = \begin{pmatrix} 1 & 3 \\ 4 & 3 \\ 1 & 0 \end{pmatrix}$ .

Nuestra nueva ecuación matricial es entonces:

$$C^T \cdot X^T = B^T \Leftrightarrow \begin{pmatrix} 0 & 1 & 1 \\ 1 & 2 & 1 \\ 2 & 2 & 0 \end{pmatrix} X^T = \begin{pmatrix} 1 & 3 \\ 4 & 3 \\ 1 & 0 \end{pmatrix}$$

Determinar el tamaño de la matriz  $X^T$  es sencillo, ya que  $X^T$  deberá tener 3 filas para poder multiplicarse por la izquierda con la matriz que nos proponen y dos columnas para que el resultado sea una matriz  $4 \times 2$ . Es decir,  $X^T$  es una matriz de incógnitas de 3 filas y 2 columnas. La definimos en SageMath:

```
Pol = PolynomialRing(Zmod(5), 6, 'x')
XT = matrix(Pol, 3, 2, Pol.gens())
```

El sistema que tenemos es por lo tanto:

$$\begin{pmatrix} 0 & 1 & 1 \\ 1 & 2 & 1 \\ 2 & 2 & 0 \end{pmatrix} \begin{pmatrix} x_0 & x_1 \\ x_2 & x_3 \\ x_4 & x_5 \end{pmatrix} = \begin{pmatrix} 1 & 3 \\ 4 & 3 \\ 1 & 0 \end{pmatrix}$$

Para resolverlo vamos a reducir la matriz  $C^T$  obteniendo su matriz de paso,  $P$ .

```

M = block_matrix([[CT,1]])
R = M.echelon_form()
P = R.subdivision(0,1)

```

$$[C^T|I] = \left( \begin{array}{ccc|ccc} 0 & 1 & 1 & 1 & 0 & 0 \\ 1 & 2 & 1 & 0 & 1 & 0 \\ 2 & 2 & 0 & 0 & 0 & 1 \end{array} \right) \rightarrow R = [N|P] = \left( \begin{array}{ccc|ccc} 1 & 0 & 4 & 0 & 4 & 1 \\ 0 & 1 & 1 & 0 & 1 & 2 \\ 0 & 0 & 0 & 1 & 4 & 3 \end{array} \right)$$

$$P = \left( \begin{array}{ccc} 0 & 4 & 1 \\ 0 & 1 & 2 \\ 1 & 4 & 3 \end{array} \right)$$

Multiplicamos ahora  $C^T$  y  $B^T$  -por la izquierda- por la matriz de paso  $P$ :

```

PCT = P*CT
PBT = P*BT

```

lo que nos da las matrices:

$$P \cdot C^T = \left( \begin{array}{ccc} 0 & 4 & 1 \\ 0 & 1 & 2 \\ 1 & 4 & 3 \end{array} \right) \left( \begin{array}{ccc} 0 & 1 & 1 \\ 1 & 2 & 1 \\ 2 & 2 & 0 \end{array} \right) = \left( \begin{array}{ccc} 1 & 0 & 4 \\ 0 & 1 & 1 \\ 0 & 0 & 0 \end{array} \right)$$

y

$$P \cdot B^T = \left( \begin{array}{ccc} 0 & 4 & 1 \\ 0 & 1 & 2 \\ 1 & 4 & 3 \end{array} \right) \left( \begin{array}{cc} 1 & 3 \\ 4 & 3 \\ 1 & 0 \end{array} \right) = \left( \begin{array}{cc} 2 & 2 \\ 1 & 3 \\ 0 & 0 \end{array} \right)$$

```

PCT=copy(PCT)
PCT.subdivide(2,[])
PBT=copy(PBT)
PBT.subdivide(2,[])

```

De esta forma el sistema  $C^T X^T = B^T$  es equivalente a:

$$P \cdot C^T \cdot X^T = P \cdot B^T \leftrightarrow \left( \begin{array}{ccc} 1 & 0 & 4 \\ 0 & 1 & 1 \\ 0 & 0 & 0 \end{array} \right) \left( \begin{array}{cc} x_0 & x_1 \\ x_2 & x_3 \\ x_4 & x_5 \end{array} \right) = \left( \begin{array}{cc} 2 & 2 \\ 1 & 3 \\ 0 & 0 \end{array} \right)$$

Como las últimas filas (tanto la de  $P \cdot C^T$  como la de  $P \cdot B^T$ ) son nulas, estas no aportan nada al sistema de ecuaciones. Por tanto, podemos quedarnos con el trozo del sistema que nos interesa:

```

N0 = PCT.subdivision(0,0)
M0 = PBT.subdivision(0,0)

```

Obteniendo el sistema equivalente:

$$\left( \begin{array}{ccc} 1 & 0 & 4 \\ 0 & 1 & 1 \end{array} \right) \left( \begin{array}{cc} x_0 & x_1 \\ x_2 & x_3 \\ x_4 & x_5 \end{array} \right) = \left( \begin{array}{cc} 2 & 2 \\ 1 & 3 \end{array} \right)$$

A la matriz reducida le falta un pivote y si lo añadimos, tenemos que añadir una nueva fila a la parte derecha de la ecuación, que debemos rellenar con parámetros libres:

```

Pivotes = matrix(Zmod(5),[[0,0,1]])
Pol2.<a,b> = PolynomialRing(Zmod(5))
Parametros = matrix(Pol2,[[a,b]])
PCE = block_matrix([[N0],[Pivotes]])
PBE = block_matrix([[M0],[Parametros]])

```

$$\begin{pmatrix} 1 & 0 & 4 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} x_0 & x_1 \\ x_2 & x_3 \\ x_4 & x_5 \end{pmatrix} = \begin{pmatrix} 2 & 2 \\ 1 & 3 \\ a & b \end{pmatrix}$$

La matriz que multiplica a la matriz de incógnitas  $X$  ya es invertible porque es cuadrada, está en forma escalonada reducida, y tiene todos sus pivotes. Por tanto, ahora podemos despejar

$$\begin{aligned} X^T &= \begin{pmatrix} x_0 & x_1 \\ x_2 & x_3 \\ x_4 & x_5 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 4 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{pmatrix}^{-1} \begin{pmatrix} 2 & 2 \\ 1 & 3 \\ a & b \end{pmatrix} = \\ &= \begin{pmatrix} 1 & 0 & 1 \\ 0 & 1 & 4 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} 2 & 2 \\ 1 & 3 \\ a & b \end{pmatrix} = \begin{pmatrix} a+2 & b+2 \\ -a+1 & -b+3 \\ a & b \end{pmatrix} \end{aligned}$$

donde  $a, b \in \mathbb{Z}_5$ .

Se sigue que la solución de la ecuación matricial original

$$XC = B \leftrightarrow \begin{pmatrix} x_0 & x_2 & x_4 \\ x_1 & x_3 & x_5 \end{pmatrix} \begin{pmatrix} 0 & 1 & 2 \\ 1 & 2 & 2 \\ 1 & 1 & 0 \end{pmatrix} = \begin{pmatrix} 1 & 4 & 1 \\ 3 & 3 & 0 \end{pmatrix}$$

es la siguiente:

$$X = \begin{pmatrix} x_0 & x_2 & x_4 \\ x_1 & x_3 & x_5 \end{pmatrix} = \begin{pmatrix} a+2 & b+2 \\ -a+1 & -b+3 \\ a & b \end{pmatrix}^T = \begin{pmatrix} a+2 & -a+1 & a \\ b+2 & -b+3 & b \end{pmatrix}$$

donde  $a, b \in \mathbb{Z}_5$ .

**Ejercicio 4.** *Calcula todas las matrices  $X$  sobre  $\mathbb{Z}_5$  tales que  $AXC = B$  siendo*

```
A = matrix(Zmod(5),[[2,1],
[2,1],
[2,1],
[3,4]])
B = matrix(Zmod(5),[[4,3,2],
[4,3,2],
[4,3,2],
[1,2,3]])
C = matrix(Zmod(5),[[4,3,2],
[1,3,2],
[4,1,4],
[2,3,2],
[4,3,2]])
```

$$A = \begin{pmatrix} 2 & 1 \\ 2 & 1 \\ 2 & 1 \\ 3 & 4 \end{pmatrix} \quad C = \begin{pmatrix} 4 & 3 & 2 \\ 1 & 3 & 2 \\ 4 & 1 & 4 \\ 2 & 3 & 2 \\ 4 & 3 & 2 \end{pmatrix} \quad B = \begin{pmatrix} 4 & 3 & 2 \\ 4 & 3 & 2 \\ 4 & 3 & 2 \\ 1 & 2 & 3 \end{pmatrix}$$

*Solución:*

Empecemos determinando el tamaño de  $X$ . Puesto que  $X$  se tiene que poder multiplicar por la derecha por  $A$  deberá tener 2 filas y como se tiene que poder multiplicar por la izquierda por  $C$  tendrá que tener 5 columnas.

```
Pol = PolynomialRing(Zmod(5),10,'x')
X = matrix(Pol,2,5,Pol.gens())
```

La ecuación nos queda:

$$\begin{pmatrix} 2 & 1 \\ 2 & 1 \\ 2 & 1 \\ 3 & 4 \end{pmatrix} \begin{pmatrix} x_0 & x_1 & x_2 & x_3 & x_4 \\ x_5 & x_6 & x_7 & x_8 & x_9 \end{pmatrix} \begin{pmatrix} 4 & 3 & 2 \\ 1 & 3 & 2 \\ 4 & 1 & 4 \\ 2 & 3 & 2 \\ 4 & 3 & 2 \end{pmatrix} = \begin{pmatrix} 4 & 3 & 2 \\ 4 & 3 & 2 \\ 4 & 3 & 2 \\ 1 & 2 & 3 \end{pmatrix}$$

Empecemos despejando la parte izquierda. Para ello hacemos la reducción

```
M1 = block_matrix([[A,1]])
R1 = M1.echelon_form()
P1 = R1.subdivision(0,1)
P1A = P1*A
P1B = P1*B
P1A.subdivide(1,[])
P1B.subdivide(1,[])
```

$$\begin{pmatrix} 1 & 3 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \end{pmatrix} \begin{pmatrix} x_0 & x_1 & x_2 & x_3 & x_4 \\ x_5 & x_6 & x_7 & x_8 & x_9 \end{pmatrix} \begin{pmatrix} 4 & 3 & 2 \\ 1 & 3 & 2 \\ 4 & 1 & 4 \\ 2 & 3 & 2 \\ 4 & 3 & 2 \end{pmatrix} = \begin{pmatrix} 2 & 4 & 1 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

Tenemos tres filas de ceros en la primera matriz que se corresponderán con tres filas de ceros en la parte derecha. Como esas filas están efectivamente en la parte derecha, podemos eliminar



ambos bloques de ceros porque no aportan ecuaciones al sistema. (Si alguna de las entradas de las tres últimas filas de la matriz de la derecha fuera no nula, el sistema sería incompatible y habríamos acabado)

```
P1A0 = P1A.subdivision(0,0)
```

```
P1B0 = P1B.subdivision(0,0)
```

$$\begin{pmatrix} 1 & 3 \end{pmatrix} \begin{pmatrix} x_0 & x_1 & x_2 & x_3 & x_4 \\ x_5 & x_6 & x_7 & x_8 & x_9 \end{pmatrix} \begin{pmatrix} 4 & 3 & 2 \\ 1 & 3 & 2 \\ 4 & 1 & 4 \\ 2 & 3 & 2 \\ 4 & 3 & 2 \end{pmatrix} = \begin{pmatrix} 2 & 4 & 1 \end{pmatrix}$$

Al igual que hemos simplificado la matriz  $A$  aplicando Gauss, podemos hacer lo propio con la matriz  $C$ . Como ella está multiplicando por la derecha a  $X$ , debemos comenzar tomando traspuertas en toda la ecuación:

$$\begin{pmatrix} 4 & 1 & 4 & 2 & 4 \\ 3 & 3 & 1 & 3 & 3 \\ 2 & 2 & 4 & 2 & 2 \end{pmatrix} \begin{pmatrix} x_0 & x_5 \\ x_1 & x_6 \\ x_2 & x_7 \\ x_3 & x_8 \\ x_4 & x_9 \end{pmatrix} \begin{pmatrix} 1 \\ 3 \end{pmatrix} = \begin{pmatrix} 2 \\ 4 \\ 1 \end{pmatrix}$$

Repetiremos el proceso hecho anteriormente, reduciendo la matriz de la izquierda:

```
A = C.T
```

```
B = P1B0.T
```

```
C = P1A0.T
```

```
M1 = block_matrix([[A,1]])
```

```
R1 = M1.echelon_form()
```

```
P1 = R1.subdivision(0,1)
```

```
P1A = P1*A
```

```
P1B = P1*B
```

```
P1A.subdivide(2,[])
```

```
P1B.subdivide(2,[])
```

$$\begin{pmatrix} 1 & 0 & 4 & 2 & 1 \\ 0 & 1 & 3 & 4 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix} \begin{pmatrix} x_0 & x_5 \\ x_1 & x_6 \\ x_2 & x_7 \\ x_3 & x_8 \\ x_4 & x_9 \end{pmatrix} \begin{pmatrix} 1 \\ 3 \end{pmatrix} = \begin{pmatrix} 3 \\ 0 \\ 0 \end{pmatrix}$$

Como podemos observar, en este caso también nos sale una fila de ceros en la parte inferior de ambos miembros (si no fuera así el sistema sería incompatible). Vamos a eliminar esta información redundante y dejar la ecuación del siguiente modo:

```
P1A0 = P1A.subdivision(0,0)
```

```
P1B0 = P1B.subdivision(0,0)
```

$$\begin{pmatrix} 1 & 0 & 4 & 2 & 1 \\ 0 & 1 & 3 & 4 & 0 \end{pmatrix} \begin{pmatrix} x_0 & x_5 \\ x_1 & x_6 \\ x_2 & x_7 \\ x_3 & x_8 \\ x_4 & x_9 \end{pmatrix} \begin{pmatrix} 1 \\ 3 \end{pmatrix} = \begin{pmatrix} 3 \\ 0 \end{pmatrix}$$

Para añadir los pivotes que nos faltan en la parte izquierda necesitamos tres filas nuevas y para añadir los pivotes en la parte derecha necesitaríamos una columna nueva. Eso hace que tengamos que convertir la matriz de la parte derecha en una matriz de 5 filas y 2 columnas usando parámetros libres. El número de parámetros es  $10 - 2 = 8$  parámetros.

```
Pivotes1 = matrix(Zmod(5), [[0,0,1,0,0], [0,0,0,1,0], [0,0,0,0,1]])
Pivotes2 = matrix(Zmod(5), [[0], [1]])
Pol2.<a,b,c,d,e,f,g,h> = PolynomialRing(Zmod(5))
BB = matrix(Pol2, [[P1B0[0,0],a],
                  [P1B0[1,0],b],
                  [c,d],
                  [e,f],
                  [g,h]])
BB.subdivide(2,1)
P1AOE = block_matrix([[P1A0], [Pivotes1]])
CE = block_matrix([[C, Pivotes2]])
Sol = (P1AOE^-1 * BB * CE^-1).T
```

$$\begin{pmatrix} 1 & 0 & 4 & 2 & 1 \\ 0 & 1 & 3 & 4 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} x_0 & x_5 \\ x_1 & x_6 \\ x_2 & x_7 \\ x_3 & x_8 \\ x_4 & x_9 \end{pmatrix} \left( \begin{array}{c|c} 1 & 0 \\ 3 & 1 \end{array} \right) = \left( \begin{array}{c|c} 3 & a \\ 0 & b \\ c & d \\ e & f \\ g & h \end{array} \right)$$

De ahí podemos ya despejar el valor de  $X$  que será

$$\begin{pmatrix} x_0 & x_5 \\ x_1 & x_6 \\ x_2 & x_7 \\ x_3 & x_8 \\ x_4 & x_9 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 4 & 2 & 1 \\ 0 & 1 & 3 & 4 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix}^{-1} \begin{pmatrix} 3 & a \\ 0 & b \\ c & d \\ e & f \\ g & h \end{pmatrix} \left( \begin{array}{c|c} 1 & 0 \\ 3 & 1 \end{array} \right)^{-1} \\ = \begin{pmatrix} 2a + c + 2d + 3e + f - g + 3h + 3 & a + d + 3f - h \\ 2b + 2c - d + e + 2f & b + 2d + f \\ c + 2d & d \\ e + 2f & f \\ g + 2h & h \end{pmatrix}$$

donde  $a, b, c, d, e, f, g$  y  $h$  son parámetros libres (toman valores cualesquiera en  $\mathbb{Z}_5$ ). Tomando traspuestas nos sale:

$$X = \begin{pmatrix} x_0 & x_1 & x_2 & x_3 & x_4 \\ x_5 & x_6 & x_7 & x_8 & x_9 \end{pmatrix} = \begin{pmatrix} 2a + c + 2d + 3e + f - g + 3h + 3 & 2b + 2c - d + e + 2f & c + 2d & e + 2f & g + 2h \\ a + d + 3f - h & b + 2d + f & d & f & h \end{pmatrix}.$$